

Ch04 流程控制-選擇結構

Ch04 內容

4.2 關係運算子

4.3 邏輯運算子

4.4 If...Then...Else 雙重選擇

4.5 If...ElseIf...Else 多重選擇

4.6 Select Case 多重選擇


4.7 巢狀選擇結構

4.8 IIf、Choose與Switch 函數

4.9 GoTo 敘述

選擇結構

- 根據**條件**來做出不同的**選擇**，此種架構在程式設計中就稱為「**選擇結構**」。
- 例如檢查A2儲存格的成績，如果成績 ≥ 60 ，就設B2儲存格值為“及格”；否則為“不及格”。



- 選擇結構會
後會回到共

述區段，執行過

4.2 關係運算子

- 關係運算子也稱**比較運算子**
- 六種關係運算子(=、<>、<、>、<= 和 >=)。
- 兩個運算元使用關係運算子來做比較，將運算後的結果傳回。其語法如下：

語法：

結果 = 運算元 1 關係運算子 運算元 2

4.2 關係運算子

- 比較結果以布林值(Boolean)傳回。
- 條件成立傳回True；不成立傳回False。
- 運算元可以為常值、變數或是運算式。

關係運算子	關係運算式	數學表示式	簡例	結果
= (相等)	$X = Y$	$X = Y$	$5 = 4 + 1$	True
<> (不等於)	$X <> Y$	$X \neq Y$	"1" <> "—"	True
< (小於)	$X < Y$	$X < Y$	"a" < "A"	False
> (大於)	$X > Y$	$X > Y$	1999/12/31 > 2000/1/1	False
<= (小於等於)	$X \leq Y$	$X \leq Y$	$4 + 2 \leq 7 - 2$	False
>= (大於等於)	$X \geq Y$	$X \geq Y$	"AC" >= "AB"	True

4.2 關係運算子

- **時間和日期**可以視為數值資料，當兩個日期做比較時，較晚的日期會大於較前的日期。
- **算術運算子的優先**次序較關係運算子高，
 - $4 + 2 \leq 7 - 2$ 結果為False。
- 可做**字串**的比較，字串是以**ASCII碼**的大小來比較。若字串第一個字元的ASCII碼相同，則再比第二字元的ASCII碼大小，依此類推。
- ASCII碼的**數字**字元ASCII碼最小，接著是**大寫字母**，然後是**小寫字母**，最後是**中文字**。

Like運算子

- 字串的比較除了使用關係運算子外，若要做到更有彈性的比對，在程式中可以使用相似程度Like運算子來比較兩個字串。
- 其語法如下：

語法：

```
結果 = string Like pattern
```

Like運算子

- string是字串型別的資料
- Like 運算子會將string和pattern（模板字串）做比較，其結果以Boolean值傳回。
- pattern是用來做比較的**模板字串**，可用**字元**、**萬用字元**或**字串**來組合成比對字串。
- pattern中常使用的字元：

模板(pattern)中的字元	string 中的符合字元
*	不限字元長度的字串（包含零個字元）
?	任意一個字元
#	任意一個數字字元(0-9)
[字串]	字串中的任一個字元
[!字串]	字串外的任何字元

Like運算子

- 模板內的 `[]` 中括號內的字串可以是逐一系列，或用連接號 (`-`) 來表示一個範圍。
- 例如 `[A-Z]` 代表A~Z廿六個字母。
- 指定字元範圍必須由小而大，例如 `[Z-A]` 是無效。
- 例
 1. `"a" Like "A" >> False`
 2. `"basic" Like "b*c" >> True`
 3. `"basic" Like "b???c" >> True`
 4. `"520" Like "###" >> True`
 5. `"s" Like "[basic]" >> True`

4.3 邏輯運算子

- 有Not、And、Or、Xor四種
- 可將多個關係運算式，組合成較複雜的邏輯運算式，來查詢多項條件是否同時成立。
- 邏輯運算式的結果有True（真）或False（假）兩種。
- 邏輯運算式的語法如下：

語法：

結果 = 運算元 A 邏輯運算子 運算元 B

優先	邏輯運算子	意義	邏輯運算式	說明	A	B	結果
1	Not	非	Not A	若 A 為真，結果為假； 若 A 為假，則結果為真。	真		假
					假		真
2	And	且	A And B	And 是當 A、B 皆為真時， 結果才為真。	真	真	真
					真	假	假
					假	真	假
					假	假	假
3	Or	或	A Or B	Or 是若 A、B 其中只要有一 個為真，結果就為真。	真	真	真
					真	假	真
					假	真	真
4	<u>Xor</u>	互斥或	A <u>Xor</u> B	A、B 中必須要有一個為真 而且一個為假，結果才為 真。	假	假	假
					真	真	假
					真	假	真
					假	真	真
					假	假	假

● 例

(score > 59) **And** (score <= 80)


(score < 0) **Or** (score > 100)

4.4 If ...Then ...Else雙重選擇

- If ... Then ... Else 雙重選擇敘述，流向只有兩種選擇。
- 例如當 <條件式> 為真時，就執行敘述區段A；不滿足<條件式> 時，執行敘述區段B。

語法

```
If <條件式> Then  
    敘述區段 A  
Else  
    敘述區段 B  
End If
```



If ...Then ...Else雙重選擇


- 若If ... Then ... Else 敘述語法，當 <條件式> 不滿足時不執行任何敘述，可省略Else部分，變成「單一選擇」。
- 語法如下：

語法

If <條件式> Then

敘述區段

End If





實作

FileName：Off.xlsm


如果 A2 儲存格的消費金額大於 2000，就在 B2 儲存格填入九折後的實付金額；否則就填入不打折的消費金額。


	A	B	C
1	消費金額	實付金額	計算
2	4000	3600	

	A	B	C
1	消費金額	實付金額	計算
2	1600	1600	

► 解題技巧

- 在CommandButton1_Click事件程序中撰寫程式。
- 當A2儲存格的值大於2000時，B2儲存格填入A2儲存格的值*0.9；否則就填入A2儲存格的值。
- 兩個不同結果要選擇，所以使用**If ... Then ... Else** 雙重選擇敘述。





Step 3 編寫程式碼

FileName: off.xlsm (工作表 1 程式碼)

```
01 Private Sub CommandButton1_Click()  
02     If Range("A2").Value > 2000 Then  
03         Range("B2").Value = Range("A2").Value * 0.9  
04     Else  
05         Range("B2").Value = Range("A2").Value  
06     End If  
07 End Sub
```


► 隨堂測驗

如果 A2 儲存格的內容為「女」，就在 B2 儲存格填入「小姐」；否則就填入「先生」等稱謂。

	A	B	C
1	性別	稱謂	確定
2	女	小姐	


	A	B	C
1	性別	稱謂	確定
2	男	先生	

4.5 If...ElseIf...Else多重選擇

- 當程式的流程需要兩個以上不同 < 條件式 > 作判斷，而且 < 條件式 > 是由上而下逐一檢查時，就可以使用到If... ElseIf ...Else多重選擇敘述。

語法：

```
If <條件式 1> Then  
    敘述區段 A  
ElseIf <條件式 2> Then  
    敘述區段 B  
ElseIf <條件式 3> Then  
    敘述區段 C  
    ⋮  
Else  
    敘述區段 N  
End If
```





實作

FileName : Great.xlsm

電腦產生兩個 1 到 99 隨機亂數，分別置入 A2 和 B2 儲存格，試比較兩個數大小並顯示結果。

	A	B	C	D
1	第一個數	第二個數	結果	
2	81	71	第一數較大	比較

	A	B	C	D
1	第一個數	第二個數	結果	
2	5	41	第二數較大	比較


► 解題技巧

- 使用Rnd內建函數時，會產生0~1之間的單精確度亂數。
- 要產生介於1~99的亂數，要配合Fix取整數的內建函數。
- 要產生最小值~最大值範圍內的整數亂數值，其公式為：

$$\text{Fix}((\text{最大值} - \text{最小值} + 1) * \text{Rnd}()) + \text{最小值}$$

► 解題技巧


- 利用亂數公式產生兩個1~99的亂數，分別存放在num1和num2兩個整數變數中。
- 兩數比較大小，會有大於、小於和等於三種情形，所以要用If...ElseIf...Else多重選擇敘述來判斷，根據條件來執行不同的結果。



Step 3 編寫程式碼

FileName: Great.xlsm (工作表 1 程式碼)

```
01 Private Sub CommandButton1_Click()  
02     Dim num1 As Integer  
03     num1 = Fix((99 - 1 + 1) * Rnd()) + 1  
04     Range("A2").Value = num1  
05     Dim num2 As Integer  
06     num2 = Fix((99 - 1 + 1) * Rnd()) + 1  
07     Range("B2").Value = num2  
08     If num1 > num2 Then  
09         Range("C2").Value = "第一數較大"  
10     ElseIf num1 < num2 Then  
11         Range("C2").Value = "第二數較大"  
12     Else  
13         Range("C2").Value = "兩數相同"  
14     End If  
15 End Sub
```



► 隨堂測驗

試使用 If...ElseIf...Else 敘述設計台鐵票別查詢程式，身高大於等於 150 公分為成人票；身高大於等於 115 公分為孩童票；低於 115 公分則免票。



	A	B	C
1	身高(公分) 票別		查詢
2	175	成人票	

	A	B	C
1	身高(公分) 票別		查詢
2	140	孩童票	

	A	B	C
1	身高(公分) 票別		查詢
2	90	免票	



4.6 Select Case 多重選擇

- 當多個條件的資料型別一樣時，可以用 **Select Case 敘述**。語法如下：



- 根據Select Case後面**運算式的結果**，**由上而下比較Case的條件值**，若符合某個Case的條件值，就會執行該敘述區段，然後跳到接在End Select後面的敘述。
- 當所有Case條件值都不滿足時，則執行**Case Else**內的敘述區段。
- 簡例

1. Case 100 :
2. Case 2, 6, 8 :
3. Case "N" , "n" :
4. Case 90 **To** 100 :
5. Case **Is** < 60 :





FileName : Prize.xlsm

設計一個獎學金試算程式，輸入分數後根據分數顯示獎學金金額。90~100 分獎學金 1000 元、80~89 獎學金 500 元、其他則顯示「請再加油！」。若分數超過 100 或低於 0 時，就顯示「分數超出範圍！」。


	A	B	C
1	成績	獎學金(元)	查詢
2	86	500元	

	A	B	C
1	成績	獎學金(元)	查詢
2	72	請再加油！	

	A	B	C
1	成績	獎學金(元)	查詢
2	120	分數超出範圍！	

► 解題技巧


- 分數為數值，有多個數值條件範圍，適合使用 Select Case 多重選擇敘述。



Step 3 編寫程式碼

FileName: Prize.xlsm (工作表 1 程式碼)

```
01 Private Sub CommandButton1_Click()  
02     Select Case Range("A2").Value  
03         Case Is > 100, Is < 0  
04             Range("B2").Value = "分數超出範圍！"  
05         Case 90 To 100  
06             Range("B2").Value = "1000 元"  
07         Case 80 To 89  
08             Range("B2").Value = "500 元"  
09         Case Else  
10             Range("B2").Value = "請再加油！"  
11     End Select  
12 End Sub
```



► 隨堂測驗

設計一個分數等級程式，使用者輸入分數可以顯示等級：優(100~90 分)、甲(89~80 分)、乙(79~70 分)、丙(69~60 分)、丁(59~0 分)，若超出範圍就顯示「分數錯誤」。

	A	B	C
1	成績	等級	查詢
2	98	優	


	A	B	C
1	成績	等級	查詢
2	72	乙	

	A	B	C
1	成績	等級	查詢
2	59	丁	

	A	B	C
1	成績	等級	查詢
2	101	分數錯誤	

4.7 巢狀選擇結構

- 若選擇結構的敘述區段內，又有選擇結構便形成「**巢狀結構**」。例如判斷三個數字何者最大：





實作

FileName：Member.xlsm

設計一個商店結帳程式，輸入消費金額和是否為會員，會根據條件顯示應付的金額。

- ① 若是會員消費滿三千元打八折；否則就打九折。
- ② 若不是會員消費滿五千元打九折；否則不打折。

► 輸出要求

	A	B	C	D
1	消費金額	會員(Y/N)	實付金額	計算
2	10000	Y	8000	

	A	B	C	D
1	消費金額	會員(Y/N)	實付金額	計算
2	2000	y	1800	


	A	B	C	D
1	消費金額	會員(Y/N)	實付金額	計算
2	7000	N	6300	

	A	B	C	D
1	消費金額	會員(Y/N)	實付金額	計算
2	4560	N	4560	

► 解題技巧

- 有多個條件要符合，必須使用巢狀選擇結構來設計。
- 巢狀結構的最外層是If...Then...Else雙重選擇結構，用Like運算子判斷B2儲存格內容是否為 "Y" 或 "y"，來區分成會員和非會員兩部分。
- 巢狀結構的內層分別用If...Then...Else雙重選擇結構，視A2消費金額的多寡來決定打折的折數，並計算出實付金額到C2儲存格。
- 使用Format()來設定輸出的格式，第二個參數為 "0" 表四捨五入到整數。

Step 3 編寫程式碼



FileName: Member.xlsm (工作表 1 程式碼)

```
01 Private Sub CommandButton1_Click()  
02     If Range("B2").Value Like "[Yy]" Then '若會員字串為 Y 或 y  
03         If Range("A2").Value >= 3000 Then '消費金額>=三千元  
04             Range("C2").Value = Format(Range("A2").Value * 0.8, "0")  
05         Else  
06             Range("C2").Value = Format(Range("A2").Value * 0.9, "0")  
07         End If  
08     Else '非會員  
09         If Range("A2").Value >= 5000 Then '消費金額>=五千元  
10             Range("C2").Value = Format(Range("A2").Value * 0.9, "0")  
11         Else  
12             Range("C2").Value = Range("A2").Value  
13         End If  
14     End If  
15 End Sub
```

► 隨堂測驗

設計一個商店結帳程式，輸入是否為會員、消費金額和是否為假日，會根據條件顯示應付的金額。

- ① 若是會員消費滿一萬元打八折；消費滿五千元打八五折；未滿五千元如果是非假日打八五折否則就打九折。
- ② 若不是會員消費滿一萬元打九折；未滿一萬元如果是非假日打九五折否則不打折。

	A	B	C	D	E
1	消費金額	會員(Y/N)	假日(Y/N)	實付金額	計算
2	12000	Y	Y	9600	

	A	B	C	D	E
1	消費金額	會員(Y/N)	假日(Y/N)	實付金額	計算
2	8000	Y	N	6800	

	A	B	C	D	E
1	消費金額	會員(Y/N)	假日(Y/N)	實付金額	計算
2	12000	N	N	10800	

	A	B	C	D	E
1	消費金額	會員(Y/N)	假日(Y/N)	實付金額	計算
2	8000	N	Y	8000	

4.8 Iif、Choose與Switch函數

Iif函數：

- If...Then...Else選擇敘述，可以改用Iif函數。
- Iif函數內的三個引數都不可省略，當第一個引數條件運算式結果為True，會傳回第二個引數；若結果為False，則傳回第三個引數。

語法：

Iif (條件運算式, True 的傳回值, False 的傳回值)

● 例：

Range("B2").Value = Iif(Range("A2").Value >= 60, "及格", "不及格")

► 隨堂測驗

請將前面 4.4 節的 Off.xlsm 實作，改使用 IIf 函數來完成。

	A	B	C
1	消費金額	實付金額	計算
2	4000	3600	

- Iif函數也可以使用**巢狀結構**，也就是**傳回值中可以使用Iif函數**，如此一來Iif函數的功能將更加強大。
- 例：台鐵票別的依據如下：
 - 身高大於等於150公分為成人票；
 - 身高大於等於115公分為孩童票；
 - 低於115公分則免票。

```
Range("B2").Value = Iif(Range("A2").Value >= 150, "成人票", _  
                        Iif(Range("A2").Value >= 115, "孩童票", "免票"))
```

► 隨堂測驗

請將前面 4.5 節的 Great.xlsm 實作，改使用巢狀 If 函數來完成。

	A	B	C	D
1	第一個數	第二個數	結果	
2	81	71	第一數較大	比較

Choose函數：

- Choose函數屬多重選擇結構，Choose函數會根據第一個引數的值(整數)，傳回相對的引數值。
- 若Choose函數的Index引數 = 1時，函數傳回值為V1；
= 2傳回V2 值，以此類推最多可到V254。
- 若Index的值小於1或大於n時，傳回值將為Null。

語法：

```
Choose(Index, V1[, V2,...[,Vn]])
```

● 例

```
Range("B2").Value = Choose(Range("A2").Value, "壹", "貳", "參", "肆", "伍")
```

Switch函數：

- Switch函數是多重選擇結構，Switch函數會根據運算式的值，傳回對應的引數值。
- Switch函數會先判斷 <運算式1>，若為真就傳回V1；否則再判斷 <運算式2>，依此類推。
- 若所有運算式的結果皆為假，則傳回Nothing。

語法：

```
Switch(運算式 1, V1[, 運算式 2, V2,... [,運算式 n, Vn]])
```

● 例

```
Range("B2").Value = Switch(zip = 220, "板橋市", zip = 300, "新竹市", _  
zip = 600, "嘉義市")
```


► 隨堂測驗

請將前面 4.6 節的 Prize.xlsm 實作，改使用 Switch 函數來完成。

	A	B	C
1	成績	獎學金(元)	查詢
2	86	500元	

4.9 GoTo敘述

- GoTo敘述可改變程式執行方向，執行GoTo敘述時，會跳到標籤(標記)名稱後面的敘述。
- 標籤名稱須符合命名規則，而且不可以重複。
- GoTo敘述只能跳到同一程序的標籤位置。
- Goto敘述不符合模組化，使用過多會難除錯。

語法：

GoTo 標籤名稱

⋮
標籤名稱： ← 注意要加冒號：





實作

FileName : GoTo.xlsm

若分數(A2 儲存格)沒有輸入資料，就以 MsgBox 顯示「沒有輸入分數」訊息。若有輸入分數，就以 MsgBox 顯示所輸入的分數。

► 輸出要求



► 解題技巧

- 當沒有輸入資料就會執行GoTo No_Keyin，直接跳到No_Keyin:標籤，後面的第05行程式不會被執行。
- 注意程式碼第06行的Exit Sub敘述，是會跳離CommandButton1_Click程序。如果不加上此行敘述，則後面的第08行敘述會被執行造成錯誤。


Step 3 編寫程式碼

FileName: Member.xlsm (工作表1 程式碼)

```
01 Private Sub CommandButton1_Click()  
02     If Range("A2").Value = "" Then  
03         GoTo No_Keyin  
04     End If  
05     MsgBox ("你的分數是 " & Range("A2").Value)  
06     Exit Sub  
07 No_Keyin:  
08     MsgBox ("沒有輸入分數")  
09 End Sub
```

► 隨堂測驗

請將上面的實作，使用 GoTo 敘述增加一個分數超出範圍的提示訊息。



On Error GoTo敘述：

- 當程式執行產生錯誤時，就必須做適當的處理才不會造成執行錯誤。
- VBA提供**On Error敘述**來處理程式的錯誤。

語法：

1. On Error Resume Next
2. On Error GoTo 標籤名稱
3. On Error GoTo 0

- **On Error Resume Next**：忽略錯誤繼續執行。
- **On Error GoTo**：錯誤發生時跳到指定的標籤。
- **On Error GoTo 0**：關閉目前程序的錯誤處理。



實作 FileName : OnError.xlsm

使用者可以輸入兩個數值，按下「相加」鈕就將結果顯示在 C2 儲存格。
如果使用者輸入非數值時，會以 MsgBox 顯示「請輸入數值」訊息。

► 輸出要求

	A	B	C	D
1	第一個數	第一個數	等於	
2	a	12		相加
3				
4				
5				
6				

Microsoft Excel


×

請輸入數值

確定

► 解題技巧

- 若使用者在A2輸入「a」，程式執行時會產生錯誤，因為把字串指定給整數變數a。




► 解題技巧

Step 3 編寫程式碼

- 在第3行加入**On Error GoTo Wrong**敘述來做錯誤處理，若產生錯誤就跳到「Wrong：」標籤。
- 第7行的**On Error GoTo 0**敘述，其作用是關閉On Error GoTo Wrong錯誤處理敘述。

FileName: Member.xlsm (工作表 1 程式碼)

```
01 Private Sub CommandButton1_Click()  
02     Dim a As Integer, b As Integer  
03     On Error GoTo Wrong  
04     a = Range("A2").Value  
05     b = Range("B2").Value  
06     Range("C2").Value = a + b  
07     On Error GoTo 0  
08     Exit Sub  
09 Wrong:  
10     MsgBox ("請輸入數值")  
11 End Sub
```



► 隨堂測驗

請將上面的實作改使用 On Error Resume Next 敘述，觀察執行的結果。

	A	B	C	D
1	第一個數 第一個數 等於			相加
2	a	12	12	

雖然不會產生錯誤，
但結果怪怪的。



FileName : Exchange.xlsm

使用者可以輸入所購買外匯的金額、幣別(必須符合 H 欄中的資料)、匯率，按下 **更新匯率** 鈕就會下載台銀匯率的資料到 H1 儲存格。「目前匯率」中可以取得指定幣別的匯率，計算出目前的報酬(台幣)，並在 B3 寫入更新的日期。如果當天已經更新過資料，就不再下載台銀匯率資料。

► 輸出要求

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
1	金額	幣別	買時匯率	目前匯率	報酬(台幣)	更新匯率		幣別	現金匯率		即期匯率		遠期匯率	歷史匯率
2	1000	美金 (USD)	32.54	32.71	170			買入	賣出	買入	賣出	買入/賣出		
3	更新日期：	2016/3/14						美金 (USD)	32.31	32.852	32.61	32.71	查詢	查詢
4								港幣 (HKD)	4.058	4.253	4.178	4.238	查詢	查詢
5								英鎊 (GBP)	45.91	47.84	46.78	47.2	查詢	查詢
6								澳幣 (AUD)	24.43	25.00	24.62	24.95	未始	未始

► 解題技巧

- 新增活頁簿並以「Exchange」為新活頁簿名稱。
- 在工作表1中建立如下表格，和ActiveX命令按鈕控制項：

	A	B	C	D	E	F	G
1	金額	幣別	買時匯率	目前匯率	報酬(台幣)	更新匯率	
2	1000	美金 (USD)	32.54				
3	更新日期：						


公式：VLOOKUP(B2,\$H\$3:\$L\$21,5,FALSE)

公式：=(D2-C2)*A2

► 解題技巧

- 因外匯匯率須上網抓取，所以可以用Excel的匯入外部資料功能，並用錄製巨集將程式碼錄下：

- ① 複製台銀匯率的網址。
- ② 指定匯入外部資料的放置儲存格。
- ③ 按 錄製巨集 鈕開始錄製巨集。



► 解題技巧

- ④ 在「資料」索引標籤中按 從Web 鈕，會開啟「新增Web查詢」視窗，設定匯入的資料範圍。



► 解題技巧

⑤ 修改外部資料範圍內容。

⑥ 停止錄製巨集。



► 解題技巧

- 將巨集1中錄製的程式碼複製，貼到 CommandButton1_Click 程序中。
- 其中.CommandType = 0 敘述必須刪除，否則會造成程式執行錯誤。
- 目前的日期可以使用 **Date** 函數取得，如果 B3 儲存格的值不等於目前時間，表示匯率資料為舊就進行更新，此時應該使用 If ... Else 選擇結構。
- 在專案視窗中的 Module1 上按右鍵，執行【移除 Module1】指令 **刪除錄製的巨集程式**。

Step 3 編寫程式碼

FileName: Exchange.xlsm (工作表 1 程式碼)

01 Private Sub CommandButton1_Click()

02 If Range("B3").Value <> Date Then '如果更新日期不等於目前日期

03 Range("B3").Value = Date '寫入更新日期

04 With ActiveSheet.QueryTables.Add(Connection:= _

05 "URL;http://rate.bot.com.tw/Pages/Static/UIP003.zh-TW.htm", Destination:= _

06 Range("\$H\$1"))

07 .CommandType = 0 '注意本行要刪除或改成註解

08 .Name = "UIP003.zh-TW_1"

09 .FieldNames = True

10 .RowNumbers = False

11 .FillAdjacentFormulas = False

12 .PreserveFormatting = True

13 .RefreshOnFileOpen = False

14 .BackgroundQuery = True

15	.RefreshStyle = xlOverwriteCells
16	.SavePassword = False
17	.SaveData = True
18	.AdjustColumnWidth = True
19	.RefreshPeriod = 0
20	.WebSelectionType = xlSpecifiedTables
21	.WebFormatting = xlWebFormattingNone
22	.WebTables = "7"
23	.WebPreFormattedTextToColumns = True
24	.WebConsecutiveDelimitersAsOne = True
25	.WebSingleBlockTextImport = False
26	.WebDisableDateRecognition = False
27	.WebDisableRedirections = False
28	.Refresh BackgroundQuery:=False
29	End With
30	Else
31	MsgBox ("已經更新")
32	End If
33	End Sub

► 隨堂測驗

請將上面的實作增加兩種外匯資料。

	A	B	C	D	E	F	G
1	金額	幣別	買時匯率	目前匯率	報酬(台幣)	更新匯率	
2	1000	美金 (USD)	32.54	32.765	225		
3	50000	日圓 (JPY)	0.31	0.2894	-1030		
4	2000	歐元 (EUR)	48.54	36.72	-23640		
5	更新日期：	2016/3/14					