

CH6 陣列的運用

陣列的運用

6.1 前言

6.2 陣列的宣告及存取

6.3 多維陣列


6.4 陣列元素值與Excel儲存格

6.1 前言

- VBA 提供「**陣列**」(**Array**)資料型別，可以將**同性質的資料**集中存放在連續的記憶體位址上。
- 例如宣告陣列名稱為score的整數陣列來存放成績，score(0) ~ score(4) 代表五位同學成績。
- score(0) ~ score(4) 為「**陣列元素**」，小括號內的數字稱為「**索引**」或「註標」，只要改變索引值便可存取陣列中的任何一個陣列元素。

```
Dim score(4) As Integer, sSum As Integer    ' Sum 為工作表函數名稱，前加 s 以示區別
score(0) = 98 : score(1) = 85 : score(2) = 76
score(3) = 60 : score(4) = 90
For i = 0 To 4
    sSum = sSum + score(i)
Next
```

- 陣列就如同一排依序排列的箱子，
- 陣列經過Dim宣告後，數量和大小就**固定**了。
- 每個箱子只能放一個指定型別的資料，若要存取只要**指定號碼**就能儲存或讀取該箱子的資料。



6.2 陣列的宣告及存取

- 6.2.1 陣列的宣告及初值設定
- 陣列必須經過宣告後才可以使用，因為陣列在宣告的同時，編譯器會依照宣告的資料型別和元素數量，會配置連續記憶體位址給該陣列。
- 陣列宣告的語法如下：

語法：

```
Dim 陣列名稱(索引 1[,索引 2[...]]) [As 資料型別]
```

6.2.1 陣列的宣告及初值設定

- 需要處理多筆**相同性質的資料**時，利用陣列中的陣列元素來取代變數是最佳的選擇。
- 陣列必須**經過宣告**後才可以使用，因為陣列在宣告的同時，編譯器會依照宣告的資料型別和元素數量，會配置**連續記憶體位址**給該陣列。
- 陣列宣告的語法如下：

語法：

```
Dim 陣列名稱(索引 1[,索引 2[...]]) [As 資料型別]
```

6.2.1 陣列的宣告及初值設定

- 陣列名稱必須遵循變數命名規則。
- 索引(註標)代表陣列的上界，**必須為整數資料**。
陣列**下界值預設為0**，**上界值加1則為陣列的大小**。
- 若只有一個索引的陣列稱為「**一維陣列**」，有兩個索引稱為「**二維陣列**」，依此類推。
- 資料型別應視存放資料的內容而定，只要夠用即可，若宣告成太大的資料型別會占用較多的記憶體空間。

一維陣列的宣告

例如宣告score為一個含有 5 個陣列元素的整數陣列，寫法如下：

```
Dim score(4) As Integer
```

—— 索引值 0 ~ 4，可當成座號 1 ~ 5

- score陣列含有score(0) ~ score(4) 共5個陣列元素，依序存放在連續記憶體位址中。
- 陣列元素存放的資料必須為整數，我們可以使用score(0)~ score(4) 來存放1到5號同學的成績。
- 另外，若成績**含有小數**(例如98.5)時，陣列就要宣告成**單精確度**或**倍精確**的資料型別。

一維陣列的宣告

- 為了提高程式的可讀性將索引值當座號，可以宣告成 `Dim score(5)`，省略 `score(0)` 不用，只用 `score(1) ~ score(5)` 來存放 1 ~ 5 號同學成績。
- 也可在宣告陣列時的索引值改成 1 To 5 來宣告：
`Dim score(1 To 5) As Integer`
- 此時會建立 `score(1)`、`score(2) ~ score(5)` 五個陣列元素，`score(0)` 元素就被省略。
- 如果在程式宣告區使用 `Option Base 1` 宣告，也可以得到相同效果而且比較簡潔：
`Option Base 1`

一維陣列元素的初值設定

- 陣列經宣告後，會為各陣列元素**填入預設初值**，若是數值陣列各陣列元素的初值預設為**0**，字串陣列元素預設為**空字串**，物件陣列元素預設為**Nothing**。
- 可以透過**指定運算子(=)**，來設定各陣列元素的初值。例如：宣告一個陣列名稱為apple的字串陣列並設定各陣列元素的初值，寫法為：

```
Dim apple(3) As String  
apple(0) = "iPod"      :   apple(1) = "iPad"  
apple(2) = "iPhone"   :   apple(3) = "iWatch"
```

6.2.1 陣列的宣告及初值設定

- 宣告陣列時未用As來指定陣列的資料型別，系統預設資料型別為Variant自由資料型別，此時各陣列元素允許存放不同資料型別的資料：

```
Dim appleData(2)  
appleData(0) = "iPhone"  
appleData(1) = 23000  
appleData(2) = True
```

6.2.2 使用迴圈存取陣列的內容

- 陣列元素的**索引值**允許使用**整數常值**、**整數變數**或**整數運算式**，存取陣列元素的內容時可透過迴圈。
- 例：計算10位同學成績的平均分數，使用陣列配合For...Next迴圈，程式寫法如下：

```
Dim score(9) , sSum, sAvg As Integer
score(0) = 98   :score(1) = 85   :score(2) = 76   :score(3) = 60   :score(4) = 90
score(5) = 86   :score(6) = 95   :score(7) = 69   :score(8) = 96   :score(9) = 56
sSum=0
For i As Integer = 0 To 9
    sSum += score(i)
Next
sAvg = sSum / 10
```

6.2.3 For Each...Next迴圈

Continue...

- 存取陣列元素時，若陣列大小無法預知可用For Each ... Next迴圈。
- For Each...Next迴圈不用告知初值和終值，會將陣列元素依序指定給varName變數，代入到迴圈內的敘述區段去執行一次，直到陣列元素都被指定完畢才離開迴圈。

語法：

```
For Each varName In arrayName
```

```
    敘述區段
```

```
    [Exit For]
```

```
    敘述區段
```

```
Next
```

6.2.3 For Each...Next迴圈

- arrayName必須是一個陣列名稱或集合名稱。
- varName為元素變數名稱，其資料型別必須和陣列或集合的資料型別一致。
- 要提前離開迴圈，可使用 Exit For敘述。
- 簡例

```
For Each s In score
    sSum = sSum + s
Next
```

語法：

```
For Each varName In arrayName
    敘述區段
    [Exit For]
    敘述區段
Next
```

6.2.4 動態陣列

Continue...

- 宣告陣列時若不知陣列的大小，可以先宣告為**動態陣列**(Dynamic Array)。再用**ReDim**敘述來**重新宣告陣列的大小**。
- 宣告陣列為動態陣列的語法如下：

語法：

```
Dim 陣列名稱() [As 資料型別]
```

- 確定陣列大小後再用**ReDim**敘述重新宣告大小：

語法：

```
ReDim [Preserve] 陣列名稱(索引 1[,索引 2[...]])
```

6.2.4 動態陣列

- ReDim不能改變原陣列的維數。
- ReDim也不能改變原陣列的資料型別。
- ReDim陣列時如果要保留原陣列的元素值，可以使用Preserve修飾詞。
- ReDim的陣列大小如果小於原陣列，超出的部分陣列元素會被清除；如果大於原陣列，新增的部分陣列元素會填入預設值。
- 例如宣告整數動態陣列，重新宣告上界為6：
Dim a() As Integer ' 宣告a為整數的動態陣列
ReDim a(6)

6.2.5 Erase敘述

- Erase敘述可以**清除陣列元素值**，並將陣列從記憶體釋放。Erase敘述必須在程式**執行階段使用**。

語法：

```
Erase 陣列名稱 [, 陣列名稱...]
```

- 例：宣告score1與 score2陣列，然後用Erase敘述刪除，陣列元素清成預設值。

```
Dim score1(30) As Integer, score2(2, 30, 16) As Integer  
Erase score1, score2      ' 刪除陣列
```

6.2.6 IsArray 函數

- IsArray() 函數可以檢查變數**是否為陣列**，若傳回值為**True**表示是陣列；**False**表示不是陣列。

語法：

```
returnValue = IsArray(arrayName)
```

- 例

```
Dim name(4) As String: Dim score(2, 5) As Integer
Dim average As Integer
Dim returnValue1, returnValue2, returnValue3 As Boolean
returnValue1 = IsArray(name)    >> 傳回值True
returnValue2 = IsArray(score)  >> 傳回值True
returnValue3 = IsArray(average) >> 傳回值False
```



實作

FileName : ScoreAvg.xlsm

試設計一個能根據輸入的學生人數當陣列的大小，然後逐一輸入成績並計算出總分和平均的程式。如果沒有輸入學生人數或型別錯誤，會顯示錯誤提示訊息。

► 輸出要求

	A	B	C	D
1	請在B2儲存格輸入學生人數			
2	學生數：	5		執行
3	座號	成績		
4	1號	100		
5				
6				

Microsoft Excel

2號同學成績：

98

確定 取消

	A	B	C	D
1	請在B2儲存格輸入學生人數			
2	學生數：	5		執行
3	座號	成績		
4	1號	100		
5	2號	98		
6	3號	76		
7	4號	87		
8	5號	94		
9	總分：	455		
10	平均：	91		

	A	B	C	D
1	請在B2儲存格輸入學生人數			
2	學生數：	五		執行
3	座號	成績		
4				
5				
6				

Microsoft Excel

請在B2儲存格輸入學生數！

確定

► 解題技巧

- 因為無法預知學生人數，所以成績score要宣告成動態整數陣列。
- 使用InputBox方法來顯示對話方塊，讓使用者輸入資料。
- 使用For...Next迴圈，逐一接受InputBox方法輸入的學生成績，存在對應的陣列元素中，並在儲存格顯示分數。
- 最後再使用For...Next迴圈，逐一讀取陣列元素值，計算出總分和平均數。。

Step 3 編寫程式碼

FileName: ScoreAvg.xlsm (工作表 1 程式碼)

```
01 Private Sub CommandButton1_Click()  
02     Dim num As Integer '學生人數  
03     Dim score() As Integer '成績動態陣列  
04     '如果 B2 為數值，且大於 1  
05     If IsNumeric(Range("B2").Value) And Range("B2").Value >= 1 Then  
06         num = Range("B2").Value '讀取學生人數  
07         ReDim score(num - 1) '重新宣告陣列大小  
08         For i = 0 To num - 1  
09             score(i) = InputBox(i + 1 & "號同學成績：")  
10             Cells(4 + i, 1) = i + 1 & "號"  
11             Cells(4 + i, 2) = score(i)  
12         Next  
13         Dim sSum As Integer '總分  
14         sSum = 0
```

15	For i = 0 To num - 1
16	sSum = sSum + score(i)
17	Next
18	Cells(num + 4, 1) = "總分 : "
19	Cells(num + 4, 2) = sSum
20	Cells(num + 5, 1) = "平均 : "
21	Cells(num + 5, 2) = sSum / num
22	Else
23	MsgBox "請在 B2 儲存格輸入學生數！"
24	End If
25	End Sub

► 隨堂測驗

將上面實作增加輸入學生姓名，並且將陣列下界設為 1。另外，將計算總分的 For ... Next 迴圈，改用 For Each ... Next 來編寫。

	A	B	C	D	E
1	請在B2儲存格輸入學生人數			執行	
2	學生數：	3			
3	座號	姓名	成績		
4	1號	喬峰	86		
5	2號	黃蓉	100		
6	3號	張無忌	53		
7		總分：	239		
8		平均：	79.66667		

6.3 多維陣列

- 使用兩個索引值，該陣列稱為「**二維陣列**」(Two-Dimensional Array)，像**表格**的資料都可以使用二維陣列來表示。
- 使用三個索引值，則該陣列稱為「**三維陣列**」，例如多個班級的各科成績。
- 若陣列的維度在二維以上，就稱為「**多維陣列**」(Multi-Dimensional Array)。
- 陣列的元素總數即是**所有維度長度的乘積**，
例：Dim score (5, 2, 3)，score陣列的陣列元素總數為72 (6x3x4)。

6.3.1 陣列的維度

Continue...

- 每個維度的大小是由零開始算起，以座號當水平列，有四位同學需四列(0~3)；以科目當垂直欄，有三科需有三欄(0~2)，因此需使用兩個索引即維度為2的二維整數陣列來宣告，各科成績如下表：

座號	計概	程式設計	電腦動畫	
1	score(0,0)=66	score(0,1)=90	score(0,2)=91	第 0 列
2	score(1,0)=77	score(1,1)=80	score(1,2)=82	第 1 列
3	score(2,0)=88	score(2,1)=70	score(2,2)=73	第 2 列
4	score(3,0)=99	score(3,1)=60	score(3,2)=64	第 3 列
	第 0 欄	第 1 欄	第 2 欄	

二維陣列的宣告

- 譬如宣告score為二維整數陣列，來存放四位同學的三科成績，寫法如下：

```
Dim score ( 3, 2 ) As Integer
```

索引 2 : 0 ~ 2 當成第 1 ~ 3 次成績

索引 1 : 0 ~ 3 當成座號 1 ~ 4

- score(0, 0) 代表座號1號同學的計概成績。
- score(1, 1) 代表座號2號同學的程式設計成績。
- 陣列元素可當成變數來運算，例如將1號同學三科成績相加，寫法為：

$sSum = score(0, 0) + score(0, 1) + score(0, 2)$

6.3.1 陣列的維度

二維陣列的元素值設定

- score陣列元素總數為 $4 \times 3 = 12$ ，共12個陣列元素。
- 第一個索引值代表座號，第二個索引值代表科目，score(0, 0)代表是座號1號同學的計概成績為66。
- score(2, 1)代表是座號3號的程式設計成績為70。
- 設定各陣列元素的初值，設定方式如下：
 - Dim score (3, 2) As Integer
 - score(0, 0) = 66 : score(0, 1) = 90 : score(0, 2) = 91' 第0列
 - score(1, 0) = 77 : score(1, 1) = 80 : score(1, 2) = 82' 第1列
 - score(2, 0) = 88 : score(2, 1) = 70 : score(2, 2) = 73' 第2列
 - score(3, 0) = 99 : score(3, 1) = 60 : score(3, 2) = 64' 第3列

6.3.1 陣列的維度

6.3.2 陣列的上、下界

- 使用 **LBound**、**UBound** 函數可以分別取得陣列的**上、下界值**。查詢時可指定要查詢陣列的哪一個維度，**省略維數**時會預設取得**第一維度**的索引值。

語法：

下界值 = LBound(陣列名稱 [, 維數])

上界值 = UBound(陣列名稱 [, 維數])

- 例：求下列宣告陣列的上下界值。
 - Dim score1(12, 15, 6)
 - max1 = UBound(score1, 1) >> 傳回12
 - Dim score2(1 To 5, 24)
 - min1 = LBound(score2) >> 傳回1



實作

FileName : Score.sln

下表為三位同學的期中考成績表，請使用二維陣列宣告並設定元素值，再計算出每位同學的總分，最後在 Excel 工作表中顯示：

姓名	微積分	計概	英文	數學
周傑倫	70	85	99	76
蔡依琳	80	95	88	92
張慧妹	90	75	77	85

► 輸出要求

	A	B	C	D	E	F	G	H
1	姓名	微積分	計概	英文	數學	總分	<input type="button" value="執行"/>	
2	周傑倫	70	85	99	76	330		
3	蔡依琳	80	95	88	92	355		
4	張慧妹	90	75	77	85	327		

► 解題技巧

- 宣告sData陣列資料型別為Variant，就可以同時存放**字串**和**整數**資料。
- 使用For ... Next巢狀迴圈，可以逐一將二維陣列的元素值指定到儲存格。
- 用**UBound(sData, 1)**可以取得**列數**，
UBound(sData, 2)可以取得**欄數**。

Step 3 編寫程式碼

FileName: Score.xlsm (工作表1程式碼)

```
01 Private Sub CommandButton1_Click()  
02     Dim sData(1 To 4, 1 To 5) As Variant  
03     sData(1, 1)="姓名": sData(1, 2)="微積分": sData(1, 3)="計概": sData(1, 4)="英文": sData(1, 5)="數學"  
04     sData(2, 1) = "周傑倫": sData(2, 2) = 70: sData(2, 3) = 85: sData(2, 4) = 99: sData(2, 5) = 76  
05     sData(3, 1) = "蔡依琳": sData(3, 2) = 80: sData(3, 3) = 95: sData(3, 4) = 88: sData(3, 5) = 92  
06     sData(4, 1) = "張慧妹": sData(4, 2) = 90: sData(4, 3) = 75: sData(4, 4) = 77: sData(4, 5) = 85  
07     For r = 1 To UBound(sData, 1)      '0 到第一維的上界值(4) - 列  
08         For c = 1 To UBound(sData, 2)    '0 到第二維的上界值(5) - 欄  
09             Cells(r, c) = sData(r, c)  
10         Next  
11     Next  
12     Dim sSum As Integer  
13     For r = 1 To UBound(sData, 1) '加入總分  
14         If r = 1 Then  
15             Cells(r, 6) = "總分"  
16         Else  
17             sSum = 0  
18             For c = 2 To UBound(sData, 2)  
19                 sSum = sSum + sData(r, c)  
20             Next  
21             Cells(r, 6) = sSum  
22         End If  
23     Next  
24 End Sub
```

► 隨堂測驗

將上面實作的成績表修改為下圖的樣式。

	A	B	C	D	E	F	G
1	姓名	微積分	計概	英文	數學	總分	執行
2	周傑倫	70	85	99	76	330	
3	蔡依琳	80	95	88	92	355	
4	張慧妹	90	75	77	85	327	
5	平均	80	85	88	84.33		

6.4 陣列元素值與Excel儲存格

6.4.1 陣列的元素值設定

Continue...

- 一、Array函數
- VBA提供Array函數可以建立一個Variant自由資料型別陣列，並同時設定陣列的初值：

語法

Array(引數串列)

- 建立的陣列是Variant資料型別，所以引數可以為任何資料型別，引數間要用逗號(,)隔開。
- 陣列的下界值由Option Base敘述決定，沒有宣告時預設為0。

6.4.1 陣列的元素值設定

Continue...

一維陣列的元素值設定

- 利用Array函數來指定陣列元素值，會比逐一指定來得簡潔。因Array建立的陣列是Variant資料型別，所以須指定給Variant資料型別陣列。
- 例如：建立一個stuData陣列，並設定姓名、身高和體重陣列元素值。
 - `Dim stuData() As Variant`
 - `stuData = Array("周傑倫", 186, 72)`
- 例：建立week陣列並設定 "日" ~ "六" 元素值。
 - `Dim week() As Variant`
 - `week = Array("日", "一", "二", "三", "四", "五", "六")`
 - `MsgBox("周末是星期" & week(6))`

6.4.1 陣列的元素值設定

二維陣列的元素值設定

- 用Array函數指定二維陣列元素值，Array函數內第一個引數為Array(第0列元素值串列)；第二個引數為Array(第1列元素值串列)。
- 例：用Array函數指定學生成績陣列元素值：

```
Dim score () As Variant  
score = Array(Array(66, 90, 91), Array(77, 80, 82), _  
              Array(88, 70, 73), Array(99, 60, 64))
```
- 要讀取Array函數指定元素值的二維陣列時，必須改用不規則陣列的特殊方式。例如score(0,0)要改用為score(0)(0)；score(1, 2)為score(1)(2)。

6.4.2 陣列與儲存格內容指定

Continue...

陣列元素值指定給儲存格

- 利用Range物件的FormulaArray屬性，就可以將陣列的元素值填入指定的儲存格範圍，而且速度比迴圈快速，其語法為：

語法：

```
Range("儲存格範圍").FormulaArray = 動態陣列名稱
```

6.4.2 陣列與儲存格內容指定

Continue...

- 儲存格範圍大小必須等於陣列大小，而且列數和欄數也要相等。
- 如果儲存格小於陣列大小，超出的元素會忽略。
- 儲存格範圍大於陣列大小，沒有對應的儲存格會被填入#N/A錯誤值。
- 例：一維陣列填入儲存格中，寫法如下：

```
Dim score (3) As Integer  
score(0) = 90: score(1) = 91: score(2) = 92: score(3) = 93  
Range("A1:D1").FormulaArray = score
```


	score(0)	score(1)	score(2)	score(3)
1	90	91	92	93

6.4.2 陣列與儲存格內容指定

Continue...

- 例：將 3×4 二維陣列的元素值填入 "A1:E2" 儲存格範圍中，寫法如下：

```
Dim score () As Variant  
score = Array(Array(90, 91, 92, 93), Array(80, 81, 82, 83) , Array(70, 71, 72, 73))  
Range("A1:E2").FormulaArray = score
```




6.4.2 陣列與儲存格內容指定

Continue...

- 要將陣列指定給垂直儲存格須設定為**二維陣列**。

```
Dim score (3,0) As Integer  
score(0,0) = 90: score(1,0) = 91: score(2,0) = 92: score(3,0) = 93  
Range("A1:D4").FormulaArray = score
```



- 也可以用Excel的**Transpose**函數將陣列水平轉向。

```
Dim score() As Variant  
score = Array(90, 91, 92, 93)  
Range("A1:A4").FormulaArray = WorksheetFunction.Transpose(score)
```

6.4.2 陣列與儲存格內容指定

Continue...

儲存格資料設定為二維陣列的元素值

- Excel工作表的儲存格範圍內的資料，也可以設定為二維陣列的元素值，其語法為：

語法：

```
Dim 動態陣列名稱 () As Variant
```

```
動態陣列名稱 = Range("儲存格範圍").Value
```

- 動態陣列的資料型別必須是Variant。
- 儲存格值指定給陣列後，要特別注意該陣列下界為1，在讀取陣列元素值時應注意。

6.4.2 陣列與儲存格內容指定

Continue...

- 例如：將Range("A1:B3")儲存格範圍的內容指定給score陣列，寫法如下：

```
Dim score() As Variant  
score = Range("A1:B3").Value
```


	A	B		
1	1號	98	score(1, 1)	score(1, 2)
2	2號	85	score(2, 1)	score(2, 2)
3	3號	76	score(3, 1)	score(3, 2)

- 儲存格範圍的內容指定給score陣列後，score(1, 1)的元素值為"1號"而不是85應留意。

6.4.2 陣列與儲存格內容指定

- 雖然只指定一欄，陣列仍然會是**二維陣列**。
- 例：將Range("A1:A3") 內容指定給score陣列：

```
Dim score() As Variant  
score = Range("A1:A3").Value
```




- 要指定score陣列的第一個元素，必須使用score(1, 1)而不是score(1)。



FileName：Total.xlsm


實作

將下表為某公司各地區四季的營業額，請建立在 Excel 工作表中。

按  鈕會讀取儲存格內資料，並統計各分公司營業額的合計。

地區	第一季	第二季	第三季	第四季
北區	200,000	300,000	400,000	500,000
中區	250,000	350,000	450,000	550,000
南區	600,000	700,000	800,000	900,000

► 輸出要求

	A	B	C	D	E	F	G
1	地區	第一季	第二季	第三季	第四季	合計	
2	北區	200,000	300,000	400,000	500,000	1,400,000	
3	中區	250,000	350,000	450,000	550,000	1,600,000	
4	南區	600,000	700,000	800,000	900,000	3,000,000	

► 解題技巧

- 宣告money為Variant型別動態陣列，再將儲存格B2:E4分公司的各季營業額指定給money陣列。
- 宣告total為長整數陣列，用來存各分公司營業額的合計。因儲存格指定給陣列時，陣列會以1為下界，所以total宣告成1 To 3使兩者相同。
- 使用For ... Next巢狀迴圈，逐一計算各分公司營業額的合計，並將合計金額存入對應total陣列元素中。
- 利用Excel的Transpose函數，將total陣列元素值轉向填入各分公司營業額的合計儲存格。

Step 3 編寫程式碼

FileName: Total.xlsm (工作表 1 程式碼)

```
01 Private Sub CommandButton1_Click()  
02     Dim money() As Variant           '動態陣列存各分公司的各季營業額  
03     money = Range("B2:E4").Value     '儲存格的內容指定給陣列  
04     Dim total(1 To 3) As Long        '長整數陣列存各分公司營業額的合計  
05     Dim sum As Long  
06     For r = LBound(money, 1) To UBound(money, 1)  
07         sum = 0                      '預設分公司營業額的合計為 0  
08         For c = LBound(money, 2) To UBound(money, 2)  
09             sum = sum + money(r, c)  
10         Next  
11         total(r) = sum               '合計金額存入對應陣列元素中  
12     Next  
13     Range("F2:F4").FormulaArray = WorksheetFunction.Transpose(total)  
14 End Sub
```

► 隨堂測驗

延續上面實作新增各季的合計，以及全年合計總額。

	A	B	C	D	E	F	G
1	地區	第一季	第二季	第三季	第四季	合計	
2	北區	200,000	300,000	400,000	500,000	1,400,000	執行
3	中區	250,000	350,000	450,000	550,000	1,600,000	
4	南區	600,000	700,000	800,000	900,000	3,000,000	
5	合計	1,050,000	1,350,000	1,650,000	1,950,000	6,000,000	

6.4.3 Filter字串搜尋函數

Continue...

- Filter函數可以一維Array字串陣列中，**找出含有指定查詢字串**(不必相等)的陣列元素。
- 因為Filter函數會將搜尋結果傳到字串陣列中，所以使用前要先宣告一個**字串動態陣列**，來接受Filter函數的傳回結果。
- 如果搜尋不到相符的陣列元素時，傳回值為**空陣列**。其語法如下：

語法：

```
returnArray = Filter(sourceArray, match, [, include [, compare ]])
```

6.4.3 Filter 字串搜尋函數

Continue...

語法：

```
returnArray = Filter(sourceArray, match, [, include [, compare ]])
```

- **returnArray**：為一個字串動態陣列，存放搜尋到的字串。
- **sourceArray**：為一個Array型態字串陣列，代表要搜尋的字串資料來源。
- **match**：代表要**搜尋的字串**。
- **include**：代表傳回值是要選取還是排除搜尋字串，預設值為**True**表傳回包含指定字串；若為**False**表傳回不包含指定字串的陣列元素。
- **compare**：是設定比對字串的方式，參數值為：
 - vbTextCompare：大小寫英文字母**視為相同**。
 - vbBinaryCompare：大小寫英文字母會**視為不同**。

語法：

```
returnArray = Filter(sourceArray, match, [, include [, compare ]])
```

● 例

```
Dim word() As Variant
```

```
word = Array("That", "is", "A", "cake")
```

```
Dim find1() As String, find2() As String, find3() As String
```

```
find1 = Filter(word, "a", True, vbTextCompare)
```

'搜到 "That" 、 "A" 、 "cake"

```
find2 = Filter(word, "a", True, vbBinaryCompare)
```

'搜到 "That" 、 "cake"

```
find3 = Filter(word, "a", False, vbBinaryCompare)
```

'搜到 "is" 、 "A"

6.4.4 Split字串分隔函數

Continue...

- 使用Split函數可以根據間隔字元，可以將字串分割成陣列元素值，常用來**指定陣列的元素值**。

語法：

```
returnArray = Split( 字串, 間隔字元)
```

- **returnArray**：為一個字串動態陣列，存放分隔後的元素值。
- **字串**：為要分割的字串，應該包含間隔字元。
- **間隔字元**：代表分割字串的字元，省略時預設為空白字元。

- 例

```
Dim word() As String
```

```
word = Split("That is a book")
```

```
Dim score() As String
```

```
score = Split("66,77,88,99,100", ",")
```

- 因為沒有指定間隔字元所以**預設為空白字元**，word字串陣列元素值分別為"That"、"is"、"a"、"book"。
- 因為間隔字元為","，所以score字串陣列元素值分別為"66"、"77"、"88"、"99"、"100"，比逐一指定元素值程式碼較為簡潔。

6.4.5 Join字串連結函數

Continue...

- Join函數可以將字串陣列元素值，用間隔字元**連結成一個字串**，常用來**顯示陣列的元素值**。

語法：

```
string = Join(sourceArray, 間隔字元)
```

- **string**：為陣列元素值用間隔字元連結後的字串。
- **sourceArray**：為一個Array型態字串陣列，代表要連結的字串資料來源。
- **間隔字元**：代表連結陣列元素間分隔的字元，省略時預設為空白字元。

6.4.5 Join字串連結函數

- 例

```
Dim word() As Variant
```

```
word = Array("That", "is", "A", "cake")
```


```
Dim j1 As String , j2 As String
```

```
j1 = Join(word)
```

```
j2 = Join(word, ",")
```

- 因為沒有指定間隔字元所以預設為空白字元，
- j1值為"That is a book"。
- 因為間隔字元指定為","，所以j2值為
"That,is,a,book"。

實作 FileName : Split.xlsm

按  鈕會將字串資料 "姓名,性別,電話,張小成,男,04-1234567,廖美美,女,01-1234567,王阿義,男,02-1234567,何樂,男,03-1234567,歐陽莉莉,女,05-1234567,周平和,男,06-1234567"，指定給 A1:C7 儲存格。

► 輸出要求

	A	B	C	D	E
1	姓名	性別	電話		
2	張小成	男	04-1234567		
3	廖美美	女	01-1234567		
4	王阿義	男	02-1234567		
5	何樂	男	03-1234567		
6	歐陽莉莉	女	05-1234567		
7	周平和	男	06-1234567		

► 解題技巧

- 字串資料是以「,」分隔，所以可以用**Split**函數來分割字串成sAry()一維陣列元素值。
- 字串資料是每三個一組(姓名、性別、電話)，用表格存放需要3欄，列數則是資料個數除以3。
- 先宣告tAry()為動態字串陣列來存放二維資料，待陣列大小確定後用ReDim重新宣告大小。
- 使用For迴圈逐一將sAry()陣列元素值指定給tAry()，必須將**計數變數i \ 3取整數做第一維索引值**，**i Mod 3取餘數做第二維索引值**。
- 使用儲存格範圍的**FormulaArray**函數，將tAry()陣列元素值轉入指定的儲存格範圍。

Step 3 編寫程式碼

FileName: Split.xlsm (工作表1 程式碼)

```
01 Private Sub CommandButton1_Click()
02     Dim data As String      '原始字串資料
03     Dim tAry() As String    '動態字串陣列，來存放二維資料
04     data = "姓名,性別,電話,張小成,男,04-1234567,廖美美,女,01-1234567,王阿義,
    男,02-1234567,何樂,男,03-1234567,歐陽莉莉,女,05-1234567,周平和,男,06-1234567"
05     Dim sAry() As String    '動態字串陣列，來存放一維資料
06     sAry = Split(data, ",") '使用 Split 函數來分割字串成陣列
07     Dim u As Integer
08     Dim r As Integer, c As Integer
09     u = UBound(sAry) + 1    '陣列元素個數
10     ReDim tAry((u / 3) - 1, 2) As String    '重新宣告陣列大小
11     For i = 0 To u - 1    '逐一讀取 sAry() 字串陣列值
12         r = i \ 3    '第一維
13         c = i Mod 3 '第二維
14         tAry(r, c) = sAry(i)    '將 sAry 陣列值指定給 tAry
15     Next i
16     'tAry 陣列值指定給儲存格範圍
17     Range(Cells(1, 1), Cells((u / 3), 3)).FormulaArray = tAry()
18 End Sub
```



► 隨堂測驗

延續上面實作更改成，陣列元素值會先轉向後再指定給儲存格。


	A	B	C	D	E	F	G
1	姓名	張小成	廖美美	王阿義	何樂	歐陽莉莉	周平和
2	性別	男	女	男	男	女	男
3	電話	04-1234567	01-1234567	02-1234567	03-1234567	05-1234567	06-1234567
4	轉換						
5							




實作 FileName : Price.xlsm

下表為產品售價試算表，售價公式為 $=\text{ROUND}(\text{Cn} * \text{Dn} * 1.4, 0)$ 。按  鈕後，食品級產品進價加一成(四捨五入到小數一位)，售價公式中 1.4 調整成 1.5；工業級產品進價加 0.5 成，售價公式不變。

► 輸出要求

	A	B	C	D	E	F	G
1	品名	等級	重量	進價	售價		
2	小蘇打	食品級	1	20	28		
3	小蘇打	工業級	1	12	17		
4	小蘇打	食品級	5	20	140		
5	小蘇打	工業級	5	12	84		
6	小蘇打	食品級	25	20	700		

調整後

	A	B	C	D	E	F	G
1	品名	等級	重量	進價	售價		
2	小蘇打	食品級	1	22	33		
3	小蘇打	工業級	1	12.6	18		
4	小蘇打	食品級	5	22	165		
5	小蘇打	工業級	5	12.6	88		
6	小蘇打	食品級	25	22	825		

► 解題技巧

- 先將儲存格範圍的資料指定給陣列，陣列元素值經過運算後再指定回儲存格範圍，這種方式**效率**會高於逐一修改儲存格值。
- 陣列元素值如果有公式時，儲存格必須改採**[R1C1]**方式指定。
- 例如B3儲存格用[R1C1] 方式為R3C2(第三列、第二欄)，D8儲存格為R8C4。

► 解題技巧

- 使用For ... Next迴圈逐一檢查data元素值，初值為1、終值可以使用UBound函數取得。
- 如果data(r, 2) = "食品級"，就改變data(r, 4) 元素值，和data(r, 5)為"=ROUND(R" & r + 1 & "C3 * R" & r + 1 & "C4 * 1.5,0)"。
- 如果data(r, 2)不等於"食品級"，就改變data(r, 4)和data(r, 5)元素值。
- 使用FormulaArray屬性，將data陣列值指定回給A2:E16儲存格範圍。
- 由本例可以了解使用VBA可以更有彈性地使用Excel，可以大幅提高效率。

Step 3 編寫程式碼

FileName: Price.xlsm (工作表 1 程式碼)

```
01 Private Sub CommandButton1_Click()  
02     Dim data() As Variant  
03     data = Range("A2:E16").Value    '儲存格值指定給 data 陣列  
04     For r = 1 To UBound(data, 1)  
05         If data(r, 2) = "食品級" Then    '如果是食品級產品  
06             data(r, 4) = Round(data(r, 4) * 1.1, 1) '進價加 1 成，四捨五入到小數 1 位  
07             '公式改為=ROUND(C2*D2*1.5,0)  
08             data(r, 5) = "=ROUND(R" & r + 1 & "C3 * R" & r + 1 & "C4 * 1.5,0)"  
09         Else  
10             data(r, 4) = Round(data(r, 4) * 1.05, 1) '進價加 0.5 成，四捨五入到小數 1 位  
11             '公式改為=ROUND(C2*D2*1.4,0)  
12             data(r, 5) = "=ROUND(R" & r + 1 & "C3 * R" & r + 1 & "C4 * 1.4,0)"  
13         End If  
14     Next  
15     Range("A2:E16").FormulaArray = data 'data 陣列值指定給儲存格範圍  
16 End Sub
```

► 隨堂測驗

延續上面實作將售價公式增加根據重量改變售價，食品級 1 公斤包裝是*1.5，5 公斤為*1.45、25 公斤為*1.4。工業級 1 公斤包裝是*1.4，5 公斤為*1.35、25 公斤為*1.3。

E4							
	A	B	C	D	E	F	G
1	品名	等級	重量	進價	售價	調整	
2	小蘇打	食品級	1	22	33		
3	小蘇打	工業級	1	12.6	18		
4	小蘇打	食品級	5	22	160		
5	小蘇打	工業級	5	12.6	85		
6	小蘇打	食品級	25	22	770		