

Ch11 自訂表單與控制項

自訂表單與控制項

11.1 自訂表單簡介

11.2 ActiveX控制項簡介

11.3 標籤與命令按鈕控制項

11.4 Text文字方塊控制項

11.5 核取方塊與切換按鈕控制項

11.6 框架與選項按鈕控制項


11.1 自訂表單簡介

- Excel工作表萬一**操作不當**將工作表的資料誤刪或誤改公式，將造成非常嚴重的後果。
- 若設計**操作介面**供使用者操作，使用者不直接操作工作表，就可**避免錯誤發生**。
- 輸入介面親和力高**容易操作**，可**縮短學習時間**。
- 可以透過**自訂表單(UserForm)**可建立出使用者操作介面(User Interface)。
- 自訂表單簡稱**表單**就是一個容器(Container)，裡面可以放置文字方塊、清單 ... 等各種**控制項(Control)**，最後編寫各個控制項的事件程序，就完成使用者介面的製作。

11.1.1 建立自訂表單

Continue...

- 執行【插入/自訂表單】指令會建立自訂表單，在專案總管中會新增一個「表單」資料夾，資料夾有一個預設名稱為UserForm1的表單物件。



11.1.2 自訂表單的常用屬性

Continue...

- 執行【檢視/屬性視窗】指令會開啟或關閉屬性視窗，透過屬性視窗可以設定自訂表單的各種屬性。
- 屬性視窗中會列出自訂表單物件的所有屬性，屬性可以依照字母順序或性質分類來排列。
- 系統會設定好預設的屬性值，我們可以視需求在設計階段修改屬性值，或在程式執行階段讀取或設定屬性值。

11.1.2 自訂表單的常用屬性

Continue...

物件清單下拉鈕
(可切換物件)

▼ 按字母順序

(可切換物件)

▼ 按性質分類

屬性 - UserForm1

UserForm1 UserForm

字母順序 性質分類

(Name)	UserForm1
BackColor	<input type="checkbox"/> &H8000000F&
BorderColor	<input checked="" type="checkbox"/> &H80000012&
BorderStyle	0 - fmBorderStyleNone
Caption	UserForm1
Cycle	0 - fmCycleAllForms
DrawBuffer	32000
Enabled	True
Font	新細明體
ForeColor	<input checked="" type="checkbox"/> &H80000012&

屬性名稱

屬性值

屬性 - UserForm1

UserForm1 UserForm

字母順序 性質分類

ShowModal	True
位置	
Height	180
Left	0
StartPosition	1 - 所屬視窗中央
Top	0
Width	240
其它	
(Name)	UserForm1
DrawBuffer	32000

屬性名稱

屬性值

11.1.2 自訂表單的常用屬性

Continue...

1. Name 屬性

Name是最重要的屬性，用來**為表單物件命名**，以方便在程式中識別。

自訂表單預設的Name屬性是「UserForm1」以此類推，可以修改成容易識別的名稱。

通常會用**frm**加上表單的功能來命名，前置字串frm代表自訂表單物件以方便識別。

例如：用來輸入資料的自訂表單，可以命名為「**frmInput**」或「**frm輸入**」。

在程式碼中可以使用**Me敘述**來**指定目前作用的自訂表單**。

11.1.2 自訂表單的常用屬性


Continue...

2. Caption 屬性

使用Caption屬性可以設定或取得自訂表單的標題欄文字。

例如：在執行階段設定frmInput自訂表單的標題文字為「輸入」，寫法為：

```
frmInput.Caption = "輸入"
```





11.1.2 自訂表單的常用屬性

Continue...

3. BackColor 屬性

使用BackColor屬性可以設定或取得**表單的背景色**。

如左下圖先點選BackColor屬性，再按屬性值的下拉鈕。如右下圖可以選用「系統配色」或「調色盤」方式來設定表單的背景色。



11.1.2 自訂表單的常用屬性

Continue...

4. Height/Width 屬性

可以設定或取得自訂表單的**高度**和**寬度**。

設計階段可以直接拖曳表單的邊框來改變大小。

5. Left/Top 屬性

可以設定或取得表單距離螢幕左上角的**水平**和**垂直距離**。

6. MousePointer/MouseIcon 屬性

可以設定或取得自訂表單的**滑鼠游標形狀**，當 MousePointer 屬性值為 fmMousePointerCustom 時，可以使用 MouseIcon 屬性所**載入的游標圖檔**（.ico 或 .cur）。

11.1.2 自訂表單的常用屬性

Continue...

7. Picture/PictureAlignment/PictureSizeMode/ PictureTiling 屬性

- 1) 使用Picture屬性可以載入一個圖檔，作為自訂表單的背景圖。
- 2) 指定Picture屬性後，可利用PictureAlignment屬性設定背景圖的位置(預設值為fmPicture AlignmentCenter-置中)。
- 3) PictureBoxSizeMode屬性設定背景圖的大小(預設值為fmPictureBoxSizeModeClip-原大小)。
- 4) PictureTiling屬性設定背景圖是否採貼磁磚填滿方式(預設值為False-不填滿)。

11.1.2 自訂表單的常用屬性

8. ScrollBars 屬性

可以設定自訂表單的捲軸狀態，預設屬性值為 `fmScrollBarsNone`(沒有水平和垂直捲軸)。

9. SpecialEffect 屬性

可以設定自訂表單的外觀，預設屬性值為 `fmSpecialEffectFlat`(平面)。

10. StartUpPosition 屬性

可以設定自訂表單的起始位置，預設屬性值為 `1`(所屬視窗的中央)。

11.1.3 自訂表單的常用方法

Continue...

1. Show 方法

可載入並**顯示**自訂表單，其後可接0或1引數值：

1 (vbModal常數值)：預設值，所顯示的表單具**排他性**，除非關閉否則不可顯示其他表單。

0 (vbModeless常數值)：代表所顯示的表單可以和其他表單**並存**。

例如：顯示完frmInput表單才顯示frmOutput：

frmInput.Show 或 frmInput.Show 1

frmOutput.Show 或 frmOutput.Show 1

例如：同時顯示frmInput和frmOutput表單：

frmInput.Show 0

frmOutput.Show 0

11.1.3 自訂表單的常用方法

Continue...

2. Hide 方法

使用Hide方法可以**隱藏**自訂表單，但是表單仍**佔據記憶體**，使用**Show**方法就可以**重新顯示**。

例如：隱藏「frmInput」自訂表單，寫法：

```
frmInput.Hide
```

3. PrintForm 方法

使用PrintForm方法可以**列印**表單。

例如：列印「frmInput」表單寫法：

```
frmInput.PrintForm
```

11.1.3 自訂表單的常用方法

4. Repaint 方法

使用Repaint方法可以重繪表單。

例如：重繪目前作用的自訂表單，寫法：

`Me.Repaint`

5. Unload 方法

使用Unload方法可以將表單由記憶體釋放，

Unload Me敘述是將目前開啟(作用) 表單刪除。


例如：關閉「frmInput」自訂表單，寫法：

`Unload frmInput`

11.1.4 自訂表單的常用事件

Continue...

- 在專案總管的表單的快顯功能表，執行【檢視程式碼】指令，開啟該表單的**程式碼視窗**。
- 在「物件」下拉式清單中選取「UserForm」，系統會建立預設的UserForm_Click事件程序。
- 在「程序」下拉式清單中可以選取需要的事件程序名稱，此時就會建立該事件程序。



11.1.4 自訂表單的常用事件

Continue...

1. Initialize 事件

Initialize事件是在自訂表單開始顯示前首先觸動的事件，而且只會被觸動一次，通常用來設定變數、表單、控制項等的初值。

例如：在Initialize事件中，設定「frmInput」自訂表單的起始位置在所屬視窗的中央，寫法：

```
Private Sub UserForm_Initialize()  
    frmInput.StartPosition = 1  
End Sub
```

11.1.4 自訂表單的常用事件

Continue...

2. Click 事件

Click事件是滑鼠在自訂表單內**按一下**所觸動的事件，是自訂表單的**預設事件**。

例如：在Click事件中設定「frmInput」自訂表單標題欄文字為「按一下」：

```
Private Sub UserForm_Click()  
    frmInput.Caption = "按一下"  
End Sub
```

11.1.4 自訂表單的常用事件

Continue...

3. Activate 事件

當表單成為**作用表單**時，會觸動Activate事件。通常也用來設定變數或控制項的**初值**。但是和Initialize事件不同，Activate事件會被**觸動多次**。表單顯示時會先觸動Initialize事件，接著才觸動Activate事件。

例如：在Activate事件中，設定「frmInput」自訂表單的背景色為淺藍色，寫法：

```
Private Sub UserForm_Activate()  
    frmInput.BackColor = RGB(150, 255, 255)  
End Sub
```

11.1.4 自訂表單的常用事件

Continue...

4. Deactivate 事件

當自訂表單變成**非作用中表單**時就會觸動 Deactivate 事件。

例如：在 Deactivate 事件中設定「frmInput」自訂表單的標題欄文字為「隱藏」，寫法：

```
Private Sub UserForm_Deactivate()  
    frmInput.Caption = "隱藏"  
End Sub
```

11.1.4 自訂表單的常用事件

Continue...

5. QueryClose 事件

自訂表單關閉前會觸動QueryClose事件，用來確保未完成的任務，例如將資料寫入工作表。

程序有Cancel和CloseMode參數，如果希望停止關閉表單，可以設Cancel參數值為True。而CloseMode參數，是代表誰關閉表單。


例如：在QueryClose事件中，用MsgBox詢問是否確定離開表單，若按 <否> 鈕就取消關閉：

```
Private Sub UserForm_QueryClose(Cancel As Integer, _  
                                CloseMode As Integer)  
    r = MsgBox("確定要離開UserForm1嗎？", vbYesNo)  
    If r = vbNo Then Cancel = True  
End Sub
```

11.2 ActiveX控制項簡介

11.2.1 工具箱


- VBA編輯器的工具箱中有許多系統提供的控制項工具，可以建立各種控制項。



11.2.2 建立控制項的步驟

Continue...

- 在白訂表單上安置控制項，要先從工具箱中點選控制項工具，用下列幾種方式來建立控制項：
 1. 在表單上按一下，就會在滑鼠位置建立一個預設大小的控制項。
 2. 在表單上拖曳滑鼠，就會建立一個指定大小的控制項。



11.2.3 控制項的編輯


Continue...

控制項在表單上建立後，我們可以利用下列編輯的方法來安排控制項。

1. 選取控制項

用**滑鼠點選**就可以選取單一控制項，如果加上**<Shift> 鍵**就可以選取多個控制項。


也可以利用工具箱的**選取物件 工具**來框選多個控制項。



11.2.3 控制項的編輯


Continue...

2. 調整控制項

將滑鼠游標移動到控制項的邊緣，可拖曳滑鼠來調整控制項的大小。當滑鼠游標呈現時，可以拖曳控制項改變控制項的位置。

3. 控制項對齊

選取多個控制項後，執行快顯功能表【對齊】指令，由出現對齊清單中選取對齊方式。






11.2.3 控制項的編輯

Continue...

4. 控制項調整大小

選取多個控制項後，按右鍵執行快顯功能表【調整大小】中的指令，可以同時調整多個控制項的大小。




11.2.3 控制項的編輯

Continue...

5. 控制項層次

點選控制項後，按右鍵執行快顯功能表【上移一層】或【下移一層】指令，可以調整控制項間上下層的關係。



11.2.3 控制項的編輯

Continue...

6. 控制項群組

選取多個控制項後，按右鍵執行快顯功能表【建立群組】指令，可以使多個控制項成為同一群組，調整時會一起移動。



11.3 標籤與命令按鈕控制項

- 使用Label標籤控制項可以在表單上顯示文字訊息，例如：顯示執行過程和結果的相關訊息。

1. Caption屬性

Caption是標籤控制項最重要的屬性，可以用來設定在上面要顯示的文字。

例如：指定lblName標籤控制項上面顯示「周傑倫」，寫法為：

```
lblName.Caption = "周傑倫"
```

11.3.1 Label標籤控制項

Continue...

2. Font屬性

Font屬性可以設定文字顯示的樣式，在設計階段可以按屬性視窗的 鈕，在「字型」對話方塊中直接設定。也可以在執行階段使用下列敘述設定。例如：使用With敘述指定lblName標籤控制項文字樣式為標楷體、大小12、粗體字，寫法為：

```
With lblName.Font
```

```
    .Name = "標楷體"
```

```
    .Size = 12
```

```
    .Bold = True
```

```
End With
```

11.3.1 Label標籤控制項

Continue...

3. ForeColor屬性

ForeColor屬性可以設定**文字的顏色**，在設計階段可以按屬性視窗的下拉鈕直接設定。

也可以在執行階段使用下列敘述設定，例如：指定lblName標籤控制項文字的顏色為紅色：

```
lblName.ForeColor = vbRed
```

4. AutoSize屬性

AutoSize屬性可以設定控制項的大小是否會隨文字的長度**自動調整**，屬性值有False(固定，預設值)、True(自動調整)。

5. BorderStyle屬性

使用BorderStyle屬性可以設定**外框的樣式**，屬性值有：

- 0(fmBorderStyleNone沒有框線，預設值)
- 1(mBorderStyleSingle單線)。

11.3.1 Label標籤控制項

6. TextAlign屬性

使用TextAlign屬性可以設定文字的對齊方式，屬性值有：

- 1(fmTextAlignLeft靠左，預設值)、
- 2(fmTextAlignCenter置中)、
- 3(fmTextAlignRight靠右)。

7. Visible屬性

使用Visible屬性可以設定顯示或隱藏控制項，屬性值有：True(顯示，預設值)、False(隱藏)。

11.3.2 CommandButton 命令按鈕控制項 Continue...


1. Enabled 屬性

Enabled 屬性可以設定按鈕是否能有效使用，屬性值有 True(有效，預設值)、False(無效)。

例如：設定 cmdOK 命令按鈕控制項有效，cmdCancel 命令按鈕無效，寫法為：

`cmdOK.Enabled = True`

`cmdCancel.Enabled = False`



11.3.2 CommandButton 命令按鈕控制項 Continue...

2. Default/Cancel 屬性

Default 屬性可設定按 **<Enter>** 鍵是否等於按該按鈕，屬性值有 False (否，預設值)、True (是)。

Cancel 屬性可以設定按 **<Esc>** 鍵是否等於按該按鈕，屬性值有 False (否，預設值)、True (是)。

3. ControlTipText 屬性

ControlTipText 屬性可以設定滑鼠移入控制項時，所顯示的提示文字。

例如：設定 cmdOK 命令按鈕控制項的提示文字為「確定」，寫法為：

```
cmdOK.ControlTipText = "確定"
```



11.3.2 CommandButton命令按鈕控制項Continue...

4. TabIndex/TabStop 屬性

TabStop屬性可以設定按 <Tab> 鍵時，該控制項是否會駐停(取得焦點 Focus)，屬性值有 True (會，預設值)、False (不會)。

TabIndex屬性可以設定按 <Tab> 鍵時，該控制項駐停的順序。


表單內的控制項預設依照建立的先後由0開始編號，但是也可以自行修改。

另外，部分控制項無法設為駐停，例如標籤控制項。

11.3.2 CommandButton 命令按鈕控制項 Continue...

5. Picture/PicturePosition 屬性

Picture屬性可以設定在控制項中顯示的圖檔。
有圖檔時可以用**PicturePosition**屬性設定圖檔和標題的相對位置，屬性值為0~12 (預設值為7)。



6. SetFocus 方法

可使得控制項成為目前駐停焦點的作用中物件，
插入點游標會移到此控制項的上面。

例如：設定cmdOK命令按鈕控制項取得駐停焦點，寫法為：

cmdOK.SetFocus

11.3.2 CommandButton命令按鈕控制項Continue...

7. Click事件

當使用者用滑鼠在控制項上**按一下**，就會觸動該控制項的Click事件。

在命令按鈕的Click事件中，我們通常會撰寫該按鈕要執行的程式碼。

Click事件是命令按鈕控制項的**預設事件**，編輯時在控制項上**快按兩下**，系統就會自動建立該控制項的預設事件程序。

11.3.2 CommandButton命令按鈕控制項

8. DbClick事件

當使用者用滑鼠在控制項上**快按兩下**，就會觸動該控制項的DbClick事件。

在命令按鈕的DbClick事件中，我們會撰寫快按兩下該按鈕時執行的程式碼。

要特別注意的是觸動DbClick事件時，**會先觸動Click事件**，所以要注意兩個程序執行的先後次序。



實作 FileName : UserForm.xlsm

在工作表按 鈕會開啟自訂表單，開啟表單時會設定物件的初值。在表單上按一下，表單上會顯示「按了表單一下!」的提示文字。按下表單的 鈕會關閉表單，但是在關閉前會先出現 MsgBox 詢問是否確定關閉？當按 <是> 鈕時才會真正關閉表單。

► 輸出要求

	A	B
1		
2		

開啟表單

自訂表單

開啟自訂表單！

關閉

自訂表單

按了表單一下！

關閉


Microsoft Excel

確定要離開嗎？

是(Y) 否(N)

► 解題技巧

- 新增一個自訂表單，在其中建立標籤和命令按鈕控制項。



- 在工作表cmdShow命令按鈕的Click事件中，使用**Show方法**開啟表單。

► 解題技巧

- 在表單的Initialize事件中，可以設定各種物件的初值。
- 在表單的Click事件中，設定lblMsg的Caption屬性值就可以改變文字。
- 在cmdUnload命令按鈕的Click事件中，使用Unload Me敘述關閉表單。
- 在表單的QueryClose事件中，用MsgBox詢問是否確定離開表單？
- 若傳回值為vbNo(表按 <否> 鈕)，就設Cancel參數值為 True來取消關閉。

Step 3 編寫程式碼

FileName: UserForm.xlsm (工作表 1 程式碼)

01 Private Sub cmdShow_Click()

02 frmTest.Show

03 End Sub

FileName: UserForm.xlsm (frmTest 程式碼)

01 Private Sub UserForm_Initialize()

02 frmTest.Caption = "自訂表單"

03 lblMsg.Caption = "開啟自訂表單！"

04 End Sub

05

06 Private Sub UserForm_Click()

07 lblMsg.Caption = "按了表單一下！"

08 End Sub

09

10 Private Sub cmdUnload_Click()

11 Unload Me

12 End Sub

13

14 Private Sub UserForm_QueryClose(Cancel As Integer, CloseMode As Integer)

15 r = MsgBox("確定要離開嗎？", vbYesNo)

16 If r = vbNo Then


17 Cancel = True

18 End If

19 End Sub

► 隨堂測驗

將上面實作增加如下的設定：



11.4 TextBox文字方塊控制項

11.4.1 TextBox文字方塊控制項常用屬性Continue...

1. Text 屬性

Text是文字方塊控制項最重要的屬性，可以用來讀取使用者輸入的**文字**，或是設定顯示的文字。例如指定msg變數等於txtNote文字方塊控制項內的文字，寫法為：

```
Dim msg As String  
msg = txtNote.Text
```

11.4.1 TextBox文字方塊控制項常用屬性Continue...

2. MaxLength 屬性

MaxLength屬性設定輸入文字的**最大長度**，預設值為**0**表示沒有限制。

例如：指定txtNote文字方塊控制項內的文字最多可以輸入5個字，寫法為：

```
txtNote.MaxLength = 5
```

3. PasswordChar 屬性

PasswordChar屬性可以設定文字方塊控制項輸入文字的**替代字元**，通常用來設計**密碼**的輸入，預設值為**""**(空字串)表示沒有替代字元。使用替代字元時，IMEMode屬性值要設為2。

11.4.1 TextBox文字方塊控制項常用屬性Continue...

4. MultiLine 屬性

MultiLine屬性設定文字方塊控制項是否**多行顯示**，屬性值有False(單行，預設值)、True(多行)。文字長度長時如果設為單行，超出範圍的文字不會顯示。

5. WordWrap屬性

當MultiLine屬性值為True(多行)時，可以再利用WordWrap屬性設定文字方塊控制項文字是否**自動換行**，屬性值有True (預設值)、False。

11.4.1 TextBox文字方塊控制項常用屬性

6. IMEMode 屬性

使用IMEMode屬性可以設定文字方塊控制項的
中文輸入法，常用屬性值：

- 0(不設定，預設值)
- 1(開啟輸入法)
- 2(關閉輸入法)

11.4.2 文字方塊控制項常用事件

Continue...

1. Change 事件

Change事件是文字方塊控制項的預設事件，當Text屬性值改變時就會觸動此事件。

通常我們會在Change事件程序中，編寫和Text屬性有關的程式，使得結果可以立即反應。

例如：在txtInch文字方塊控制項的Change事件中，指定lblCm的Caption屬性值，以達到英吋立即轉換成公分的效果，寫法為：

```
Private Sub txtInch_Change()  
    lblCm.Caption = Val(txtInch.Text) * 2.54  
End Sub
```

2. Enter 事件

當文字方塊控制項取得**駐停焦點**時，就會觸動Enter事件。

通常我們會在Enter事件程序中，編寫設定文字方塊**初值**有關的程式。

例：在txtInch文字方塊控制項的Enter事件中，指定txtInch的Text屬性值等於空字串，以達到清空內容等候輸入的效果，寫法為：

```
Private Sub txtInch_Enter()  
    txtInch.Text = ""  
End Sub
```

11.4.2 文字方塊控制項常用事件

Continue...

3. Exit 事件

當文字方塊控制項失去駐停焦點時，就會先觸動Exit事件。在Exit事件程序中，編寫檢查文字方塊輸入值有關的程式，如果檢查不合條件，就可以設Cancel參數值為True，停止離開重回文字方塊。

例如：在txtInch文字方塊控制項的Exit事件中，檢查Text屬性值若等於空字串就不離開：

```
Private Sub txtInch_Exit(ByVal Cancel As _  
    MSForms.ReturnBoolean)  
    If txtInch.Text = "" Then Cancel = True  
End Sub
```

11.4.2 文字方塊控制項常用事件

Continue...

4. KeyPress 事件

當文字方塊控制項取得駐停焦點時，使用者在鍵盤按下字元鍵就會觸動KeyPress事件。

在KeyPress事件中，編寫檢查輸入字元的程式。

透過事件的KeyAscii參數值可以取得輸入的字元，如果檢查不合條件就設KeyAscii為0，取消輸入的字元，插入點重回原輸入位置。

例如：在txtInch文字方塊控制項的KeyPress事件中，設定只能輸入0~9的數值字元：

```
Private Sub txtInch_KeyPress(ByVal KeyAscii As _  
    MSForms.ReturnInteger)  
    If KeyAscii < 48 Or KeyAscii > 57 Then KeyAscii = 0  
End Sub
```

11.4.2 文字方塊控制項常用事件

Continue...

常用按鍵的ASCII碼列表如下：

常用的按鍵	ASCII 碼值
 至 	48 至 57
 至  、  至 	65 至 90 , 97 至 122
 和 	13 和 27
 (退位鍵)	8
	32
 、  、  、  、 	46 、 43 、 45 、 42 、 47

5. KeyDown/KeyUp 事件

當文字方塊控制項取得駐停焦點時，使用者**按下按鍵**就會觸動**KeyDown**事件，**放開**時則會觸動**KeyUp**事件。

因為**任何按鍵**都會觸動KeyDown和KeyUp事件，通常我們會在該事件程序中，編寫**檢查特殊字元**的程式。

透過事件程序的**KeyCode**參數值可以取得輸入的字元，字元鍵的KeyCode值和ASCII碼相同。

11.4.2 文字方塊控制項常用事件

- **Shift參數值**可以得知 **<Shift>** 、 **<Ctrl>** 、 **<Alt>** 按鍵被按的狀態，例如0(未按)、1(按<Shift>)、2(按<Ctrl>)、3(按<Shift> + <Ctrl>)、4(按<Alt>)。
- 例如：在txtInch文字方塊控制項的 KeyKeyDown事件中，檢查如果按 鍵就顯示輸入文字，寫法為：

```
Private Sub txtInch_KeyDown(ByVal KeyCode As _  
    MSForms.ReturnInteger, ByVal Shift As Integer)  
    If KeyCode = 13 Then MsgBox "輸入資料" & _  
        txtInch.Text  
End Sub
```





實作 FileName : TextBox.xlsm

開啟活頁簿時會自動開啟表單，在表單中允許輸入使用者名稱和密碼(最多 6 個英數字元)。按 **確定** 鈕會檢查使用者名稱(Excel)和密碼(2016)是否正確？若正確就關閉表單；若不正確共可輸入三次，錯誤三次顯示訊息後就關閉活頁簿。輸入使用者後按 **Enter↵** 鍵就跳到密碼處，密碼輸入後按 **Enter↵** 鍵就跳到 **確定**。如果按 **關閉** 鈕會關閉活頁簿。

► 輸出要求

► 解題技巧

- 新增一個自訂表單，在其中建立標籤、文字方塊和兩個命令按鈕控制項。



- 因為希望一開啟活頁簿就會出現表單，所以**Show**方法要寫在活頁簿的**Open**事件中。

► 解題技巧

- 在txtUser的KeyPress事件中，檢查KeyAscii是否等於13(按Enter鍵)，若是就執行SetFocus方法將焦點(插入點游標)移到txtPassWord。
- 在txtPassWord的KeyPress事件中，檢查KeyAscii是否在48~57(數字)、65 ~ 90(大寫英文字母)和97~122(小寫英文字母)之間，否則就設KeyAscii = 0清除輸入的字元。
- 在cmdOK命令按鈕的Click事件中，檢查輸入次數是否小於3，以及使用者和密碼是否正確。因為輸入次數要能累計，所以要宣告成靜態變數。
- 如果錯誤三次，就使用ThisWorkbook.Close敘述關閉活頁簿。

Step 3 編寫程式碼

FileName: TextBox.xlsm (ThisWorkbook 程式碼)

01 Private Sub Workbook_Open()

02 frmPassword.Show

03 End Sub

FileName: TextBox.xlsm (frmPassword 程式碼)

01 Private Sub UserForm_Initialize()

02 txtPassWord.IMEMode = 2 '關閉中文輸入

03 txtPassWord.PasswordChar = "*" '替代字元為*

04 txtPassWord.MaxLength = 6 '最多 6 個字元

05 txtUser.SetFocus '駐停焦點移到使用者

06 End Sub

07

08 Private Sub txtUser_KeyPress(ByVal KeyAscii As MSForms.ReturnInteger)

09 If KeyAscii = 13 Then txtPassWord.SetFocus '若按 Enter 鍵就跳到密碼

10 End Sub

11

12 Private Sub txtPassWord_KeyPress(ByVal KeyAscii As MSForms.ReturnInteger)

13 If KeyAscii = 13 Then cmdOK.SetFocus

14 Select Case KeyAscii

15 Case 48 To 57 '數字

16 Case 65 To 90 '大寫英文字母

17 Case 97 To 122 '小寫英文字母

18 Case Else

19 KeyAscii = 0 '清除輸入的字元

20 End Select

21 End Sub

22

23 Private Sub cmdOK_Click()

24	Static num As Integer	'num 要累計所以要宣告成靜態變數
25	num = num + 1	'輸入次數加 1
26	If num < 3 Then	'若輸入次數小於 3
27	If txtUser.Text = "Excel" And txtPassWord.Text = "2016" Then	'若正確
28	Unload Me	'關閉表單
29	Else	
30	MsgBox "錯誤！"	
31	txtUser.SetFocus	'駐停焦點移到使用者
32	End If	
33	Else	
34	MsgBox "錯誤三次！"	
35	ThisWorkbook.Close	'關閉活頁簿
36	End If	
37	End Sub	

38

39 Private Sub cmdUnload_Click()

40	ThisWorkbook.Close	'關閉活頁簿
41	End Sub	

► 隨堂測驗


- 將上面實作增加進入使用者和密碼文字方塊時，先將文字內容清空的功能。
- 離開輸入使用者名稱和密碼的文字方塊時，檢查文字內容不能空白。以及輸入密碼的文字方塊只能輸入數字字元的功能。

11.5 核取方塊與切換按鈕控制項

11.5.1 CheckBox核取方塊控制項

Continue...

1. Value 屬性



可以取得和設定核取方塊控制項的**核取狀態**，屬性值為False(未核取，預設值)、True(核取)。

例如若chkVIP核取方塊控制項被勾選就設off變數為0.9；否則就設為1，寫法為：

```
If chkVIP.Value = True Then
```

```
    off = 0.9
```

```
Else
```

```
    off = 1
```

```
End If
```


11.5.1 CheckBox核取方塊控制項

2. Change 事件

當核取方塊控制項的核取狀態改變時，也就是Value屬性值改變時就會觸動Change事件。

在Change事件程序中，編寫和核取狀態改變的相關程式碼。

例如在chkVIP核取方塊控制項的Change事件程序中，根據勾選狀態來設定off(折扣變數)的變數值為0.9或1，寫法為：

```
Private Sub chkVIP_Change()
```

```
    Dim off As Single
```

```
    off = 1
```

```
    If chkVIP.Value = True Then off = 0.9
```

```
End Sub
```

11.5.2 ToggleButton 切換按鈕控制項 Continue...



1. Value 屬性

使用Value屬性可以取得和設定切換按鈕控制項的**點按狀態**，屬性值為False(未點按外觀呈凸起，預設值)、True(點按外觀呈凹陷)。

例如：若tgbSex切換按鈕控制項被點按就設切換按鈕的文字為「女」；否則就設為「男」，寫法為：

```
If tgbSex.Value = True Then  
    tgbSex.Caption = "女"  
Else  
    tgbSex.Caption = "男"  
End If
```

11.5.2 ToggleButton 切換按鈕控制項

2. Change 事件

當切換按鈕控制項的點按狀態有改變，也就是 Value 屬性改變時就會觸動 Change 事件。

我們可以在 Change 事件程序中編寫和點按狀態改變的相關程式碼。



實作 FileName : CheckBox.xlsm

按工作表的 **輸入資料** 鈕會開啟自訂表單，使用者可以輸入姓名、電話和勾選是否參加。按 **輸入** 鈕如果有輸入姓名，就將資料寫入工作表中，並將資料清除等待輸入。按下表單的 **關閉** 鈕會關閉表單。

► 輸出要求

The figure illustrates the output requirements for the 'CheckBox.xlsm' file, showing the data entry process and the resulting spreadsheet state.

Top Screenshot (Form Open):

- The spreadsheet shows columns A (姓名), B (參加), and C (電話). The '輸入資料' button is highlighted with a callout 1.
- The custom form '資料' is open, showing input fields for 姓名 (張三丰), 參加 (checked), and 電話 ((02)1234567). The '輸入' button is highlighted with a callout 5, and the '關閉' button is highlighted with a callout 4.

Bottom Screenshot (Form Closed):

- The spreadsheet shows the data entered into cells A1, B1, and C1: 姓名 (張三丰), 參加 (參加), and 電話 ((02)1234567). The '輸入資料' button is highlighted with a callout 6.
- The custom form '資料輸入' is closed, showing empty input fields for 姓名, 參加, and 電話.

► 解題技巧

- 在工作表1中建立下面表格，和一個ActiveX命令按鈕控制項，並命名為cmdInput。

	A	B	C	D	E
1	姓名	參加	電話	輸入資料	
2	儲存格格式為文字				

- 新增一個自訂表單，在其中建立標籤、文字方塊、核取方塊和命令按鈕控制項。

UserForm:
Name = frmInput

TextBox:
Name = txtName

CheckBox:
Name = chkJoin
Caption = "參加"

TextBox:
Name = txtTel

CommandButton:
Name = cmdOK

CommandButton:
Name = cmdUnload

資料輸入

姓名：

參加：

電話：

輸入

關閉

► 解題技巧

- 在cmdOK命令按鈕的Click事件中，檢查txtName的Text屬性值是否為空白，若不是空白就將資料寫入。
- 使用UsedRange.Rows.Count + 1敘述，來取得工作表空白儲存格的第一列。
- 將表單中控制項的值，分別指定給工作表中對應的儲存格，以達成資料寫入工作表的效果。

Step 3 編寫程式碼

FileName: CheckBox.xlsm (工作表 1 程式碼)

01 Private Sub cmdInput_Click()

02 frmInput.Show

03 End Sub

FileName: CheckBox.xlsm (frmInput 程式碼)

01 Private Sub cmdOK_Click()

02 If txtName.Text <> "" Then

03 r = Sheets(1).UsedRange.Rows.Count + 1 '空白的第一列

04 Cells(r, 1) = txtName.Text

05 Cells(r, 2) = IIf(chkJoin.Value, "參加", "不參加") '使用 IIf 函數指定字串值

06 Cells(r, 3) = txtTel.Text

07 End If

08 txtName.Text = ""

09 chkJoin.Value = False

10 txtTel.Text = ""

11 txtName.SetFocus

12 End Sub

13

14 Private Sub cmdUnload_Click()

15 Unload Me

16 End Sub

► 隨堂測驗

將上面實作的核取方塊改為切換按鈕控制項，當點按參加時才可以輸入人數資料。

	A	B	C	D	E
1	姓名	參加	人數	電話	輸入資料
2	張三丰	參加		(04)1234567	
3	李四川	不參加		0900-000000	
4					
5					
6					
7					
8					
9					
10					

資料輸入

姓名：

參加：

人數：

電話：

切換按鈕

點選參加才可輸入

11.6 框架與選項按鈕控制項


- 使用Frame框架控制項可以將表單內的控制項**分門別類**放置，搬移時會一起動作方便編輯。
- 更重要的功能是可以**自成一個群組**，和外面控制項隔離不會相互影響。
- 編輯時要先建立框架控制項，然後在其中拖曳出控制項，該控制項就屬於框架的子物件。

1. Caption屬性

使用Caption屬性可以取得和設定框架控制項的**標題文字**，作為同一類控制項的標題。

11.6.2 OptionButton選項按鈕控制項 Continue...

- 當有多個選項**只能選擇其一**的時候，就可以使用OptionButton選項按鈕控制項。
- 選項按鈕具**互斥性**，也就是說一組選項按鈕中只能一個被選取，如果另一個選項按鈕被選取時，其他的選項按鈕會自動設為未選取。
- 若希望表單中有多組性質不同的選項按鈕時，就必須使用**框架控制項來做區隔**。



11.6.2 OptionButton選項按鈕控制項 Continue...

1. Alignment 屬性

Alignment屬性可以設定選項鈕和標題文字的相對位置，屬性值為1(選項鈕在標題文字的右邊，預設值)、0(選項鈕在左邊)。

2. Value 屬性

使用Value屬性可以取得和設定選項按鈕控制項的被選取的狀態，屬性值為False(未被選取，預設值)、True(被選取)。

例如：若optMale選項按鈕控制項被選取就設str字串變數為「先生」，寫法為：

```
If optMale.Value = True Then str = "先生"
```

11.6.2 OptionButton選項按鈕控制項 Continue...

3. **GroupName** 屬性

除了使用框架控制項來區隔選項按鈕外，也可以用GroupName屬性來設定選項按鈕**所屬的群組**。

例如：設optMale和optFemale選項按鈕同屬grpA群組(群組名稱可自訂)，寫法為：

```
optMale.GroupName = "grpA"  
optFemale.GroupName = "grpA"
```

11.6.2 OptionButton選項按鈕控制項 Continue...

3. Click 事件

當在選項按鈕控制項上按一下會觸動Click事件的點按狀態改變，因為該選項按鈕會被設為選取狀態，所以可以在Click事件程序中，編寫選取該選項按鈕時的相關程式碼。

例如：若optRed選項按鈕控制項被按一下時，就就設lblMsg標籤控制項的文字為紅色：

```
Private Sub optRed_Click()  
    lblMsg.ForeColor = vbRed  
End Sub
```

11.6.2 OptionButton選項按鈕控制項

4. Change 事件

當選項按鈕控制項的選取狀態改變，也就是Value屬性改變時就會觸動Change事件。

當選項按鈕數量較多時，我們也可以在按鈕控制項的Click事件程序中，逐一檢查選項按鈕的選取狀態。



實作 FileName : OptionButton.xlsm

按工作表的 **開啟表單** 鈕會開啟自訂表單，工作表中的資料會顯示在表單中。按 **下一筆** 鈕會顯示下一筆資料，到最後一筆時該鈕不能使用。按 **前一筆** 鈕會顯示前一筆資料，到第一筆時該鈕不能使用。按下表單的 **關閉** 鈕會關閉表單。

► 輸出要求

編號	姓名	性別	本薪	參加團保
1	廖美昭	女	74000	加保
2	林姍姍	女	56400	不加保
3	張志成	男	38800	加保
4	王婷瑋	女	21200	加保
5	何佳儀	男	43600	加保
6	吳明哲	男	64000	不加保
7	蔡和平	男	31600	不加保
8	蘇如意	女	49200	加保

開啟表單

UserForm1 ×
編號： 3 性別： ☒ 男
姓名： ☐ 女
本薪： 團保： ☒ 參加

前一筆 **下一筆** **關閉**

► 解題技巧

- 新增一個自訂表單，在其中建立標籤、文字方塊、選項按鈕、核取方塊和命令按鈕控制項。

UserForm:
Name = frmData

Label:
Name = lblNO

TextBox:
Name = txtName

TextBox:
Name = txtSalary

CommandButton:
Name = cmdForward

UserForm1

編號: Label6

姓名:

本薪:

性別: ☐ 男 ☐ 女

團保: ☐ 參加

前一筆

下一筆

關閉

OptionButton:
Name = optMale

OptionButton:
Name = optFemale

CheckBox:
Name = chkJoin

CommandButton:
Name = cmdUnload

CommandButton:
Name = cmdNext

The image shows a Visual Basic UserForm titled 'UserForm1'. It contains several controls: a label '編號:' (ID) with text 'Label6', a text box '姓名:' (Name), another text box '本薪:' (Basic Salary), three radio buttons for '性別:' (Gender) with options '男' (Male) and '女' (Female), a checkbox '團保:' (Insurance) with option '參加' (Participate), and three command buttons: '前一筆' (Previous), '下一筆' (Next), and '關閉' (Close). Arrows point from the control names and their names (e.g., Name = frmData, Name = lblNO, Name = txtName, Name = txtSalary, Name = cmdForward, Name = optMale, Name = optFemale, Name = chkJoin, Name = cmdUnload, Name = cmdNext) to the corresponding controls in the form.

► 解題技巧

- 因為顯示第幾筆資料和共有多少筆資料會在多個程序中共用，所以宣告成公用的表單成員變數no和max。
- 在表單的Initialize事件程序中設定no和max的初值，並從工作表中將第1筆資料讀入，顯示在對應的控制項當中。因為讀取資料的程式在多個程序中都需要，所以獨立成一個Sub程序GetData以供呼叫。

► 解題技巧

- GetData程序內的參數是指第幾筆資料，會到指定的儲存格讀取資料，並轉到表單對應的控制項中顯示。
- 在cmdForward_Click事件程序中設定no-1，如果no=1就設Enabled = False，使前一筆 鈕不能使用。然後呼叫GetData程序讀取資料。
- 在cmdNext_Click事件程序中設定no+1，如果no=max就設Enabled = False，使下一筆 鈕不能使用。然後呼叫GetData程序讀取資料。

Step 3 編寫程式碼

FileName: OptionButton.xlsm (工作表 1 程式碼)

01 Private Sub cmdShow_Click()

02 frmInput.Show

03 End Sub

FileName: OptionButton.xlsm (frmData 程式碼)

01 Dim no As Integer, max As Integer

02 Private Sub UserForm_Initialize()

03 no = 2 '預設為第 2 筆

04 max = Sheets(1).UsedRange.Rows.Count - 1 '取得資料總筆數

05 cmdForward_Click '呼叫 cmdForward_Click 程序到第 1 筆資料

06 End Sub

07

08 Private Sub cmdForward_Click()

09 no = no - 1 '筆數減 1

10 If no = 1 Then cmdForward.Enabled = False '前一筆鈕不能使用

11 cmdNext.Enabled = True '下一筆鈕可以使用

12 GetData (no) '讀取資料

13 End Sub

14

15 Private Sub cmdNext_Click()

16 no = no + 1 '筆數加 1

17 If no = max Then cmdNext.Enabled = False '下一筆鈕不能使用

18 cmdForward.Enabled = True '前一筆鈕可以使用

19 GetData (no)

20 End Sub

21

22 Sub GetData(ByVal n As Integer)

23 Dim ws As Worksheet

24 Set ws = Sheets(1)

25 Dim rng As Range

26 Set rng = ws.Cells(n + 1, 1) '筆數對應的儲存格

27 lblNO.Caption = rng.Value '讀取編號

28 txtName.Text = rng.Offset(0, 1).Value '右移一格讀取姓名

29 If rng.Offset(0, 2).Value = "男" Then '右移兩格如果是 男

30 optMale.Value = True '選取男選項按鈕

31 Else


32 optFemale.Value = True '選取女選項按鈕

33 End If



34 txtSalary.Text = rng.Offset(0, 3).Value '右移三格

35	If rng.Offset(0, 4).Value = "加保" Then '右移四格如果是 加保
36	chkJoin.Value = True '勾選參加核取方塊
37	Else
38	chkJoin.Value = False '不勾選參加核取方塊
39	End If
40	End Sub
41	
42	Private Sub cmdUnload_Click()
43	Unload Me
44	End Sub

► 隨堂測驗

將上面實作增加一個  鈕，點按後會將表單內的資料寫入工作表中對應的儲存格中。

	A	B	C	D	E	F	G	H	I	J	K
1	編號	姓名	性別	本薪	參加團保						
2	1	廖美昭	女	74000	加保						
3	2	林姍姍	女	56400	不加保						
4	3	張志成	男	38800	加保						
5	4	王婷瑋	女	21600	不加保						
6	5	何佳儀	男	43600	加保						
7	6	吳明哲	男	64000	不加保						
8	7	蔡和平	男	31600	不加保						
9	8	蘇如意	女	49200	加保						



UserForm1

編號： 4 性別： ☐ 男

姓名： ☒ 女

本薪： 團保： ☐ 參加